July 2020



PopKAT Default Pharmacokinetic Models

Structures and Parameters

Frederic Y. Bois
INERIS, Verneuil en Halatte, France

Enhancing the reliability, efficiency, and usability of Bayesian population PBPK modeling (Grant number: 1U01FD005838-01)

Table of Contents

Summary

This report describes four pharmacokinetic models which have been implemented in the *GNU MCSim* language used by PopKat. Users may modify those models as they wish to suit their needs.

Introduction

Four models have been implemented in PopKat. They result from the combination of either a minimal distribution models or a fully physiologically-based pharmacokinetic (PBPK) distribution model with either a simple linear absorption model or a physiological advanced dissolution and absorption model (openCAT) (see Figure 1). The various models' equations and parameters are given next.



Figure 1: The four model combinations implemented so far in *PopKat*.

Units

In all following models, time is in hours, lengths are in dm, surface areas in dm^2, volumes are in liters (dm^3), masses of substances (quantities) in micromoles, and concentrations of substances in microM. For convenience with published parametric sub-models, there are some exceptions for specific parameters (for example the galenic powder density parameter is in g/ml).

Absorption models

First-order absorption

In this case, oral dosing is modeled as an input rate proportional at any time to the dose remaining to be absorbed (*i.e.*, the dose remaining in the gut lumen at time *t*):

$$Rate\,of\,input \;=\; K_a \cdot Q_{lumen} \qquad (1)$$

The rate of input at time *t* has dimension of quantity per time (µmol / h), K_a is the absorption rate constant (1/h), and Q_{lumen} is the quantity in the gut lumen at time *t* (in µmol).

Q_{lumen} needs to be modeled separately. Oral infusion can be modeled as a constant. For bolus dosing, Q_{lumen} should decrease exponentially with time, at rate K_a. *PopKat* offers the input function *PerExp* to model that case (see *GNU MCSim* documentation http://www.gnu.org/software/mcsim). *PerExp* function is parameterized by four arguments: the ingested dose (*Dose*, in µmol), the period of the exposure / no exposure cycle (*Period*, in h), an absorption lag-time (*Lag*, in h) and the absorption rate constant (K_a, in 1/h). You can, for example, simulate a 100 µmol oral dosing, once per day, with an absorption lag of half an hour and an absorption rate of 1 per hour, by setting the corresponding parameters to the needed value (see the model file and example input file provided with the software). More complex cases can be simulated, but that requires the development of specific equations, at the user's hand.

The parameters of this simple absorption model have null default values.

Advanced dissolution and absorption model (openCAT)

This gastro-intestinal tract model has six segments (stomach, duodenum, jejunum, ileum, cecum and colon), plus a liver compartment (Figure 2). Blood comes in from arteries and goes out to the liver portal vein. The liver has its own arterial blood supply and can then be linked to different distribution models, such as a two-compartment model, or a PBPK model. In any case, part of the quantity leaving the liver is recirculated to the gut walls.

Each gastro-intestinal tract segment has a lumen compartment, an epithelial layer compartment, and a underlying tissular (wall) compartment. The equations governing drug fate in each compartment and segment of the transit model are given below, together with the drug release and dissolution sub-models implemented in the version 3 and later of that model. The parameter scaling equations are given next. The openCAT model is for now only coupled to the two-compartment model.



Figure 2: Schematic representation of the *PopKat* advanced compartmental absorption and transit model (openCAT). Input blood flow irrigates the gut segment walls. The double arrows indicate the possibility for active influx and efflux transport. The openCAT model is here linked to a minimal PBPK model (in fact liver coupled to the two-compartment model).

Main differential equations

The differential equation for the quantity of drug in the liver is the sum of terms for blood transport, active transporter influx from blood, active transporter efflux from liver tissue and metabolism:

$$\frac{\partial Q_{liver}}{\partial t} = R_{in} + F_{liver} \cdot C_{input} - \left(F_{portal} + F_{liver}\right) \cdot C_{blood,liver} + \frac{V_{max,inf_{liver}} \cdot C_{u_{blood,liver}}}{K_{m,inf} + C_{u_{blood,liver}}} - \frac{V_{max,eff_{liver}} \cdot C_{u_{liver}}}{K_{m,eff} + C_{u_{liver}}} - \frac{V_{max,met_{liver}} \cdot C_{u_{liver}}}{K_{m,met} \cdot f_{u_{liver,vitro}} + C_{u_{liver}}} \quad (2)$$

where R_{in}, whose formula is given below, is the rate of input from the gut walls *via* the portal blood flow (F_{portal}); F_{liver} is the direct arterial flow to the liver; C_{input} is the input (arterial) blood drug concentration; $C_{blood,liver}$ is the blood concentration at the liver exit; $V_{max,inf_{liver}}$ is the Michaelis-Menten maximum rate of active drug transport from blood into liver tissue; $C_{u_{blood,liver}}$ is the concentration of unbound drug in blood at the liver exit; $K_{m,inf}$ is the Michaelis-Menten constant for active transport into tissues (influx) (assumed to be the same for all tissues); $V_{max,eff_{liver}}$ is the Michaelis-Menten maximum rate of active drug transport from liver tissue to blood; $C_{u_{liver}}$ is the concentration of unbound drug in liver tissue; $K_{m,eff}$ is the Michaelis-Menten constant for active transport out of

tissues (efflux) (assumed to be the same for all tissues); $V_{max,met_{liver}}$ is the Michaelis-Menten maximum rate of metabolism in liver; $K_{m,met}$ is the Michaelis-Menten constant for metabolism (assumed to be the same for all tissues); and $f_{u_{liver,vitro}}$ is the fraction of drug unbound in the *in vitro* system used to estimate liver metabolism and available for metabolism.

$C_{blood,liver}$ is given by:

$$C_{blood,liver} = \frac{C_{u_{blood,liver}}}{f_{u_{blood}}} \quad (3)$$

where $f_{u_{blood}}$ is the fraction of drug unbound in blood (computed as $f_{u_{plasma}}$, fraction unbound in blood plasma, divided by the blood over plasma concentration ratio r_{BP}).

$C_{u_{blood,liver}}$ is given by:

$$C_{u_{blood,liver}} = \frac{C_{u_{liver}}}{K_{puu_{liver}}} \quad (4)$$

where $K_{puu_{liver}}$ is the equilibrium ratio (partition coefficient) of unbound drug concentration in liver over unbound concentration in blood.

$C_{u_{liver}}$ is given by:

$$C_{u_{liver}} = f_{u_{liver}} \cdot C_{liver} \quad (5)$$

where $f_{u_{liver}}$ is the fraction of drug unbound in liver tissue, and C_{liver} is the total liver concentration (Q_{liver} divided by V_{liver}).

Note that the apparent liver over blood partition coefficient, PC_{liver}, is given by:

$$PC_{liver} = \frac{K_{puu_{liver}} \cdot f_{u_{blood}}}{f_{u_{liver}}} \quad (6)$$

Finally, R_{in} is given by:

$$R_{in} = \sum_{i \in O} F_i \cdot C_{blood,wall_i} \tag{7}$$

where the summation is taken over the set O of gut segments (O = {*stomach*, *duodenum*, *jejunum*, *ileum*, *cecum*, *colon*}), F_i is the blood flow perfusing segment i, $C_{blood,wall_i}$ is the drug blood concentration at the exit of segment i wall. Note that the sum of the flows F_i is equal to F_{portal}.

Similarly to liver, $C_{blood,wall_i}$ is given by:

$$C_{blood,wall_i} = \frac{C_{u_{blood,wall_i}}}{f_{u_{blood}}} \tag{8}$$

$C_{u_{blood,wall_i}}$ is given by:

$$C_{u_{blood,wall_i}} = \frac{C_{u_{wall_i}}}{K_{puu_i}} \tag{9}$$

where K_{puu_i} is the equilibrium ratio (partition coefficient) of unbound drug concentration in segment i wall over unbound concentration in blood.

$C_{u_{wall_i}}$ is given by:

$$C_{u_{wall_i}} = f_{u_i} \cdot C_{wall_i} \tag{10}$$

where f_{u_i} is the fraction of drug unbound in segment i tissues (same value assumed for wall and epithelium), and C_{wall_i} is the total segment i wall drug concentration (Q_{wall_i} divided by V_{wall_i}).

The differential equations governing drug quantity in gut walls for the various segments are:

$$\frac{\partial Q_{wall_i}}{\partial t} = F_i \left(C_{input} - C_{blood,wall_i}\right) + K_{a_i} \left(C_{u_{epith_i}} - K_{E/At} \cdot C_{u_{wall_i}}\right) \tag{11}$$

where C_{input} is the (arterial) blood concentration in input of segment i, K_{a_i} is the absorption flow in segment i, $C_{u_{epith_i}}$ is the free drug concentration in the epithelium of segment i, and $K_{E/At}$ is the wall to epithelium excretion over absorption ratio.

$C_{u_{epith_i}}$ is given by:

$$C_{u_{epith_i}} = f_{u_i} \cdot C_{epith_i} \tag{12}$$

where C_{epith_i} is the total segment *i* epithelium drug concentration (Q_{epith_i} divided by V_{epith_i}).

For the drug quantities in epithelia, Q_{epith_i}, the differential equations are:

$$\begin{aligned} \frac{\partial Q_{epith_i}}{\partial t} = \; & K_{a_i}\left(C_{diss_i} - K_{E/Ae} \cdot C_{u_{epith_i}}\right) - K_{a_i}\left(C_{u_{epith_i}} - K_{E/At} \cdot C_{u_{wall_i}}\right) \\ & + \frac{V_{max,inf_i} \cdot C_{diss_i}}{K_{m,inf} + C_{diss_i}} - \frac{V_{max,eff_i} \cdot C_{u_{epith_i}}}{K_{m,eff} + C_{u_{epith_i}}} \\ & - \frac{V_{max,met_i} \cdot C_{u_{epith_i}}}{K_{m,met} \cdot f_{u_{i,vitro}} + C_{u_{epith_i}}} \end{aligned} \tag{13}$$

where C_{diss_i} is the dissolved drug concentration in the lumen of segment *i*; $K_{E/Ae}$ is the epithelium to lumen excretion over absorption ratio in *i*; V_{max,inf_i} is the Michaelis-Menten maximum rate of active drug transport from lumen into epithelium in segment *i*; $K_{m,inf}$ is the Michaelis-Menten constant for active transport into tissues (influx) (assumed to be the same for all tissues); V_{max,eff_i} is the Michaelis-Menten maximum rate of active drug transport from epithelium tissue to lumen in segment *i*; $K_{m,eff}$ is the Michaelis-Menten constant for active transport out of tissues (efflux) (assumed to be the same for all tissues); V_{max,met_i} is the Michaelis-Menten maximum rate of metabolism in epithelium *i*; $K_{m,met}$ is the Michaelis-Menten constant for metabolism (assumed to be the same for all tissues); and $f_{u_{i,vitro}}$ is the fraction of drug unbound in the *in vitro* system used to estimate epithelium *i* metabolic parameters.

Drug dissolution equations

For the dissolved drug quantities in lumina, Q_{diss_i}, the differential equations are:

$$\begin{aligned} \frac{\partial Q_{diss_i}}{\partial t} = \; & R_X - K_{t_i} \cdot C_{diss_i} - K_{a_i}\left(C_{diss_i} - K_{E/Ae} \cdot C_{u_{epith_i}}\right) + K_{diss_i}(t) \cdot Q_{undiss_i} - K_{precip} \cdot Q_{diss_i} \\ & - \frac{V_{max,inf_i} \cdot C_{diss_i}}{K_{m,inf} + C_{diss_i}} + \frac{V_{max,eff_i} \cdot C_{u_{epith_i}}}{K_{m,eff} + C_{u_{epith_i}}} \end{aligned} \tag{14}$$

where R_X is the *dissolved* drug release rate in the stomach (see below), for the other gut segments R_X is zero; K_{t_i} is the intestinal transit rate in i; $K_{diss_i}(t)$ is the instantaneous drug dissolution rate in i (see next paragraph), and K_{precip} is the drug precipitation rate constant (assumed to be same in all gut segments).

The instantaneous dissolution rate constants, $K_{diss_i}(t)$ in 1/hours, are computed at any time as:

$$K_{diss_i}(t) = K_{diss}(C_{sat_i} - C_{diss_i}) \tag{15}$$

where K_{diss} is the drug baseline dissolution rate constant (in 1/microM/hours), and C_{sat_i} is the saturation concentration in segment i (see below).

For the undissolved drug quantities in lumina, Q_{undiss_i}, the differential equations are:

$$\frac{\partial Q_{undiss_i}}{\partial t} = R'_X - K_{t_i} \cdot C_{undiss_i} - K_{diss_i}(t) \cdot Q_{undiss_i} + K_{precip} \cdot Q_{diss_i} \tag{16}$$

In the stomach, R'_X is the *undissolved* drug release rate (see below). For the other g.i. tract segments, R'_X is zero.

The dissolved and undissolved drug quantities excreted in the feces are given by:

$$\frac{\partial Q_{diss_{feces}}}{\partial t} = K_{t_{colon}} \cdot C_{diss_{colon}} \tag{17}$$

$$\frac{\partial Q_{undiss_{feces}}}{\partial t} = K_{t_{colon}} \cdot C_{undiss_{colon}} \tag{18}$$

Drug Release models

Drug release in the stomach can take various forms. First, the drug can be released in a dissolved or undissolved form (which may then equilibrate according to equations 14 and 16). Next, the release can be immediate or delayed. Those four cases are triggered by four flags and are mutually exclusive.

Release in dissolved form

***Immediate release*:** This case is triggered by setting the variable $G_{immediated}$ to 1. If so, rate R_X is equal to the oral administration rate D_{rate} (a model input, in micromoles / hours). Rate R'_X for the undissolved form is set to zero

in this case. A bolus administration can be simulated by giving D_{rate} a high value for a short time, or by setting it to zero and setting the initial value of the state variable $Q_{diss_{stomach}}$ to the bolus dose prescribed.

***Delayed release*:** This case is triggered by setting the variable $G_{delayed_d}$ to 1. If so, R_X, is currently modeled by a Weibull function:

$$R_X = Dose \frac{k}{\lambda}\left(\frac{t-\tau}{\lambda}\right)^{k-1} e^{-\left(\frac{t-\tau}{\lambda}\right)^k} \quad (19)$$

The oral bolus dose, *Dose*, must be given as a model input, together with the Weibull slope *k*, scale λ and lag time τ.

An alternative form would be a first order delayed release (using it would require a small change in the model equations). In that case, we would have:

$$R_X = K_{release} \cdot Q_{remain} \quad (20)$$

which would require the following differential equation for the quantity Q_{remain} remaining to be released to the stomach at time *t*:

$$\frac{\partial Q_{remain}}{\partial t} = D_{rate} - R_X \quad (21)$$

Release in undissolved form

***Immediate release*:** This case is triggered by setting the variable $G_{immediate_u}$ to 1. If so, rate R'_X is equal to the oral administration rate D_{rate} (a model input, in micromoles / hours), and rate R_X for the dissolved form is set to zero. A bolus administration can be simulated by giving D_{rate} a high value for a short time, or by setting it to zero and setting the initial value of the state variable $Q_{undiss_{stomach}}$ to the bolus dose prescribed

***Delayed release*:** This case is triggered by setting the variable $G_{delayed_u}$ to 1. If so, R'_X is modeled by the same Weibull function as in equation 19 above, and rate R_X for the dissolved form is set to zero.

Miscellaneous equations

Miscellaneous equations are used to compute various quantities of eventual interest. The quantities absorbed in each gut segment are computed by integration of the following differential:

$$\frac{\partial Q_{abs_i}}{\partial t} = K_{a_i}\left(C_{diss_i} - K_{E/Ae}\cdot C_{u_{epith_i}}\right) + \frac{V_{max,inf_i}\cdot C_{diss_i}}{K_{m,inf} + C_{diss_i}} - \frac{V_{max,eff_i}\cdot C_{u_{epith_i}}}{K_{m,eff} + C_{u_{epith_i}}} \quad (22)$$

The quantity metabolized in each gut segment is computed by integration of the following differential:

$$\frac{\partial Q_{met_i}}{\partial t} = \frac{V_{max,met_i}\cdot C_{u_{epith_i}}}{K_{m,met}\cdot f_{u_{i,vitro}} + C_{u_{epith_i}}} \quad (23)$$

The quantity reaching the portal vein at time t ($Q_{abs_{portal}}$) has the following differential:

$$\frac{\partial Q_{abs_{portal}}}{\partial t} = R_{in} \quad (24)$$

see equation 7 for the definition of R_{in}.

The instantaneous apparent rate of gut absorption is:

$$K_a = \frac{Q_{abs_{portal}}}{\sum_{i \in O} Q_{diss_i}} \quad (25)$$

The major mass balance checking equations are:

$$Q_{diss} = \sum_{i \in O} Q_{diss_i} \quad (26)$$

$$Q_{epith} = \sum_{i \in O} Q_{epith_i} \quad (27)$$

$$Q_{wall} = \sum_{i \in O} Q_{wall_i} \quad (28)$$

$$Q_{abs} = \sum_{i \in O} Q_{abs_i} \tag{29}$$

$$Q_{elim_{gut}} = \sum_{i \in O} Q_{met_i} + Q_{diss_{feces}} + Q_{undiss_{feces}} \tag{30}$$

Parameters and their scaling

Many of the parameters used in the above equations are scaled prior to simulation, in order to simplify input and to automatically model correlations between parameters. We give here the corresponding equations. The list of primary physiological parameters (the value of which the user may set to any value for a particular simulation) is recapitulated in Table 1, together with their units and default values. The drug-specific parameters are listed in Table 2.

The volumes (in liters) of gut tissues (wall) in the various gut segments are computed as:

$$V_{wall_i} = f_{Bw_i} \cdot B_m \tag{31}$$

B_m is the body mass (kg) and f_{Bw_i} is the fraction of body mass corresponding to tissue segment *i* (see Table 1) (tissue densities are assumed to be equal to 1 throughout).

Total blood flow (cardiac output, in L/h) is assumed to depend on B_m:

$$F_{total} = sc_{F_{total}} \cdot B_m^{0.75} \tag{32}$$

The blood flows (L/h) to the various gut segments are given by:

$$F_i = f_{F_i} \cdot F_{total} \tag{33}$$

The fraction, f_{F_i}, of cardiac output flowing to tissue segment *i* is given in Table 1.

Liver portal blood flow is simply:

$$F_{portal} = \sum_{i \in O} F_i \tag{34}$$

The epithelial volume for each segment is given by:

$$V_{epith_i} = S_i \cdot H_{epith} \tag{35}$$

The gut epithelial thickness, H_{epith} (in dm), is given in Table 1. The epithelial surface area, S_i (in dm^2), for each segment is given by:

$$S_i = 2\pi R_i L_i \tag{36}$$

The radius, (R_i, in dm), and length (L_i, in dm), for each segment is given in Table 1.

The luminal volume for each segment *i* is given by:

$$V_{lumen_i} = f_{Bl_i} \cdot B_m \tag{37}$$

The fraction of body mass corresponding to each lumen segment *i* (f_{Bl_i}) is given in Table 1.

The intestinal transit rates per segment are computed as:

$$K_{t_i} = \log(2) \frac{V_{lumen_i}}{T_{1/2_i}} \tag{38}$$

The transit half lives in each segment *i* ($T_{1/2_i}$, in h) is given in Table 1.

The Michaelis-Menten maximum rate of metabolism in segment *i* (V_{max,met_i} in micromol/h), or in liver ($V_{max,met_{liver}}$) is scaled from the value measured *in vitro*, for example, using:

$$V_{max,met_i} = 60 \times 1000 \cdot V_{max_v} \cdot \mu_{p_i} \cdot V_{epith_i} \tag{39}$$

where V_{max_v} is the *in vitro* measured value (in micromol/min/mg of microsomal proteins), and μ_{p_i} is the microsomal abundance in *i* (in mg/g of tissue).

Similarly, for liver:

$$V_{max,met_{liver}} = 60 \times 1000 \cdot V_{max_v} \cdot \mu_{p_{liver}} \cdot V_{liver} \tag{40}$$

The *in vivo* Michaelis-Menten constant for metabolism in gut ($_{Km,met}$) is simply the value measured *in vitro*:

$$K_{m,met} = K_{m_v} \tag{41}$$

According to Agoram *et al*. (2001. Advanced Drug Delivery Reviews 50:S41-S67) the baseline dissolution rate constant, K_{diss} (in 1/microM/hours) is computed as:

$$K_{diss} = 1080 \times MM \times \frac{Diffusivity}{D_g . R_g . H_{diff}} \quad (42)$$

where *MM* is the drug molecular mass, *Diffusivity* is the drug diffusion coefficient, D_g is the galenic powder density (in *g/ml*) (see Table 2), R_g is the galenic powder radius (in *micrometers*) (see Table 2), and H_{diff} is the diffusion layer thickness (in *micrometers*). H_{diff} is equal to R_g if R_g is between 5 and 30 micrometers. Otherwise, it is set to the nearest bound (5 or 30 micrometers) (Arav *et al*. 2012, Drug Development and Industrial Pharmacy, 38:940-951).

The drug diffusion coefficient (in cm^2/second) at infinite dilution is estimated as a function of its molal volume (in ml/g) by the empirical equation (from Wilke & Chang, 1955, AIChE. Journal, 1:264-270):

$$Diffusivity = 1.09 \times 10^{-4} \times M_{vol}^{-0.6} \quad (43)$$

The drug molal volume has to be provided by the user.

C_{sat_i} depends on the acidic / basic character of the drug of interest (coded by the AB_{type} flag parameter) and on the pH of the i^{th} gut segment (Henderson-Hasselbach equations):

- If the drug is neutral ($AB_{type} = 0$)

$$C_{sat_i} = Solubility \quad (44)$$

- If the drug is a base ($AB_{type} = 1$)

$$C_{sat_i} = Solubility \times \left(1 + 10^{pK_b - pH_i}\right) \quad (45)$$

- If the drug is an acid ($AB_{type} = 2$)

$$C_{sat_i} = Solubility \times \left(1 + 10^{pH_i - pK_a}\right) \quad (46)$$

- If the drug is an ampholyte ($AB_{type} = 3$):

$$C_{sat_i} = Solubility \times \left(1 + 10^{pK_b - pH_i} + 10^{pH_i - pK_a}\right) \quad (47)$$

The drug *Solubility* parameter is listed in Table 2.

Finally, epithelial permeability in gut segment *i* is given by:

$$K_{a_i} = P_{eff} \cdot SA_i \cdot f_{abs_i} \quad (48)$$

where P_{eff} is the effective permeability of gut epithelia (see Table 2), SA_i the epithelial surface area of *i* (calculated from the lengths and radii given in Table 1), and f_{abs_i} a 0 or 1 flag parameter turning absorption on or off in segment *i* (see Table 2).

Table 1: Physiological parameters of the advanced compartmental absorption and transit model.

Parameters	Symbols	Units	Default values	References
Body mass	B_m	kg	70	
Cardiac output scaling coefficient	$sc_{F_{total}}$	$L.kg^{-0.75}$	15	
Fractions of cardiac output going to tissues				[1]
stomach	$f_{F_{stom}}$	-	0.024	
duodenum	$f_{F_{duod}}$	-	0.016	
jejunum	$f_{F_{jeju}}$	-	0.056	
ileum	$f_{F_{ileum}}$	-	0.033	
cecum	$f_{F_{cecum}}$	-	0.006	
colon	$f_{F_{colon}}$	-	0.038	
Tissue fractions of body mass				[1]
stomach	$f_{Bw_{stom}}$	-	0.0021	
duodenum	$f_{Bw_{duod}}$	-	0.0003	
jejunum	$f_{Bw_{jeju}}$	-	0.0009	
ileum	$f_{Bw_{ileum}}$	-	0.0006	
cecum	$f_{Bw_{cecum}}$	-	0.0005	
colon	$f_{Bw_{colon}}$	-	0.0048	
Lumina as fractions of body mass				[1]
stomach	$f_{Bl_{stom}}$	-	0.0036	
duodenum	$f_{Bl_{duod}}$	-	0.0003	
jejunum	$f_{Bl_{jeju}}$	-	0.0023	
ileum	$f_{Bl_{ileum}}$	-	0.0032	
cecum	$f_{Bl_{cecum}}$	-	0.0001	
colon	$f_{Bl_{colon}}$	-	0.0051	
Lengths of gut segments				
stomach	L_{stom}	dm	2.83	[2]
duodenum	L_{duod}	dm	1.41	[2]

Parameters	Symbols	Units	Default values	References
jejunum	L_{jeju}	dm	11.68	[2]
ileum	L_{ileum}	dm	17.52	[2]
cecum	L_{cecum}	dm	1.7	[1]
colon	L_{colon}	dm	11	[3]
Radii of gut segments				
stomach	R_{stom}	dm	0.967	[2]
duodenum	R_{duod}	dm	0.153	[2]
jejunum	R_{jeju}	dm	0.137	[2]
ileum	R_{ileum}	dm	0.098	[2]
cecum	R_{cecum}	dm	0.35	[1]
colon	R_{colon}	dm	0.25	[1]
Transit half-lives in lumina (hours)				[1]
stomach	$T_{1/2_{stom}}$	h	0.25	
duodenum	$T_{1/2_{duod}}$	h	0.25	
jejunum	$T_{1/2_{jeju}}$	h	1.02	
ileum	$T_{1/2_{ileum}}$	h	2.04	
cecum	$T_{1/2_{cecum}}$	h	4.55	
colon	$T_{1/2_{colon}}$	h	13.5	
pH of luminal contents				[1]
stomach	pH_{stom}	-	1.7	
duodenum	pH_{duod}	-	6	
jejunum	pH_{jeju}	-	6.5	
ileum	pH_{ileum}	-	7.4	
cecum	pH_{cecum}	-	5.9	
colon	pH_{colon}	-	7	
Microsomal proteins				
liver	$\mu_{P_{liver}}$	mg / g of tissue	45	[4]
stomach	$\mu_{P_{stom}}$	mg / g of tissue	0	
duodenum	$\mu_{P_{duod}}$	mg / g of tissue	18	[5]
jejunum	$\mu_{P_{jeju}}$	mg / g of tissue	25	[5]
ileum	$\mu_{P_{ileum}}$	mg / g of tissue	24	[5]
cecum	$\mu_{P_{cecum}}$	mg / g of tissue	0	
colon	$\mu_{P_{colon}}$	mg / g of tissue	0	
Gut epithelium thickness	H_{epith}	dm	0.000525	

[1]: Perdaems *et al.* Clinical Pharmacokinetics 2010; 49:239-258.

[2]: Ando *et al.* Drug Metabolism and Disposition 2015; 43:590–602.

[3]: Valentin. Annals of the ICRP 2002; 32:1-277.

[4]: Houston *et al.* Toxicology in Vitro 2012; 26:1265-1271.

[5]: Paine *et al.* The Journal of Pharmacology and Experimental Therapeutics 1997; 283:1552-1562.

Table 2: Drug-specific parameters of the advanced compartmental absorption and transit model.

Parameters	Symbols	Units	Default values
Molecular mass	M	g/mol	0
Acido-basic type	AB_{type}	-	0
Basic dissociation constant	pK_b	-	0
Acid dissociation constant	pK_a	-	0
Absorption switches			
stomach	$f_{abs_{stom}}$	-	0
duodenum	$f_{abs_{duod}}$	-	0
jejunum	$f_{abs_{jeju}}$	-	0
ileum	$f_{abs_{ileum}}$	-	0
cecum	$f_{abs_{cecum}}$	-	0
colon	$f_{abs_{colon}}$	-	0
Dosage form switches (mutually exclusive)			
immediate release, dissolved	$G_{immediate_d}$	-	1
immediate release, undissolved	$G_{immediate_u}$	-	0
delayed release, dissolved	$G_{delayed_d}$	-	0
delayed release, undissolved	$G_{delayed_u}$	-	0
Weibull delayed release parameters			
Slope	k	-	1
Scale	λ	h	1
Lag	τ	h	0
Galenic radius	R_g	microm	25
Powder density	D_g	g/ml	1.2
Intrinsic water solubility	$Solubility$	microg/L	0
Solubility factor for acids	SF_A	-	100
Solubility factor for bases	SF_B	-	50
Precipitation rate	K_{precip}	1/h	0
Excretion over absorption rate constant ratios			
between lumen and epithelium	$K_{E/Ae}$	-	0
between epithelium and tissue	$K_{E/At}$	-	0
Effective permeability of gut epithelia	P_{eff}	dm/h	0
Unbound tissue / unbound blood part. coeffs			
stomach	$K_{puu_{stom}}$	-	1
duodenum	$K_{puu_{duod}}$	-	1
jejunum	$K_{puu_{jeju}}$	-	1
ileum	$K_{puu_{ileum}}$	-	1
cecum	$K_{puu_{cecum}}$	-	1
colon	$K_{puu_{colon}}$	-	1
liver	$K_{puu_{liver}}$	-	1
Fractions unbound in tissues			
stomach	$f_{u_{stom}}$	-	1
duodenum	$f_{u_{duod}}$	-	1
jejunum	$f_{u_{jeju}}$	-	1
ileum	$f_{u_{ileum}}$	-	1
cecum	$f_{u_{cecum}}$	-	1

Parameters	Symbols	Units	Default values
colon	$f_{u_{colon}}$	-	1
liver	$f_{u_{liver}}$	-	1
Fraction unbound in plasma	$f_{u_{plasma}}$	-	1
Blood over plasma concentration ratio	r_{BP}	-	1
Fraction unbound in *in vitro* metabolic assay			
stomach	$f_{u_{stom,vitro}}$	-	1
duodenum	$f_{u_{duod,vitro}}$	-	1
jejunum	$f_{u_{jeju,vitro}}$	-	1
ileum	$f_{u_{ileum,vitro}}$	-	1
cecum	$f_{u_{cecum,vitro}}$	-	1
colon	$f_{u_{colon,vitro}}$	-	1
liver	$f_{u_{liver,vitro}}$	-	1
Metabolism *Vmax* measured *in vitro*	$V_{max,v}$	µmol/min/mg microsomal prot	0
Metabolism *Km* measured *in vitro*	$K_{m,v}$	microM	0
Maximum rates of active influx to tissues			
stomach	$V_{max,inf_{stom}}$	µmol/h	0
duodenum	$V_{max,inf_{duod}}$	µmol/h	0
jejunum	$V_{max,inf_{jeju}}$	µmol/h	0
ileum	$V_{max,inf_{ileum}}$	µmol/h	0
cecum	$V_{max,inf_{cecum}}$	µmol/h	0
colon	$V_{max,inf_{colon}}$	µmol/h	0
liver	$V_{max,inf_{liver}}$	µmol/h	0
Active influx *Km*	$K_{m,inf}$	microM	0
Maximum rates of efflux from tissues			
stomach	$V_{max,eff_{stom}}$	µmol/h	0
duodenum	$V_{max,eff_{duod}}$	µmol/h	0
jejunum	$V_{max,eff_{jeju}}$	µmol/h	0
ileum	$V_{max,eff_{ileum}}$	µmol/h	0
cecum	$V_{max,eff_{cecum}}$	µmol/h	0
colon	$V_{max,eff_{colon}}$	µmol/h	0
liver	$V_{max,eff_{liver}}$	µmol/h	0
Active efflux *Km*	$K_{m,eff}$	microM	0

Distribution and elimination models

One-compartment model

This model to do.

Two-compartment model

This model is a straightforward implementation of the classical two-compartment model with input to and elimination from the central compartment (see Figure 3).



Figure 3: Schematic representation of the two-compartment distribution model.

Main differential equations

The differential equation governing the drug quantity in the central compartment ($Q_{central}$) is:

$$\frac{\partial Q_{central}}{\partial t} = K_a \cdot D_{input} - K_e \cdot Q_{central} - K_{c\,to\,p} \cdot Q_{central} + K_{p\,to\,c} \cdot Q_{periph} \quad (40)$$

where K_a is the absorption rate constant, D_{input} is the drug quantity remaining to be absorbed at time t, K_e is the elimination rate constant, $K_{c\,to\,p}$ is the central to peripheral transport rate constant, $K_{p\,to\,c}$ is the peripheral to central transport rate constant, and Q_{periph} is the drug quantity in the peripheral compartment.

D_{input} is the computed at any time t as a periodic exponential decay function:

$$Dinput = \text{PerExp}\ (Dose,\ Period,\ Tlag,\ Ka); \quad (41)$$

Dose is the administered dose (in micromoles), *Period* is the period (in h) of the exposure / no exposure cycle, *Tlag* is the eventual absorption lag-time (in h), and K_a the absorption rate constant (1/h). In a given period, input starts at *Tlag* after the start of the period, with value *Dose*. It then decays exponentially with rate constant K_a until the end of the period, at which point it is turned off. To simulate a single dose, an arbitrarily large period (say 1 billion hours) can be used.

The differential equation for Q_{periph} is:

$$\frac{\partial Q_{periph}}{\partial t} = K_{c\ to\ p} \cdot Q_{central} - K_{p\ to\ c} \cdot Q_{periph} \tag{42}$$

The differential equation for the quantity eliminated Q_{elim} is:

$$\frac{\partial Q_{elim}}{\partial t} = K_e \cdot Q_{central} \tag{43}$$

Miscellaneous equations

Two miscellaneous equations are used to compute concentrations in each compartment:

$$C_{central} = \frac{Q_{central}}{V_{central}} \tag{44}$$

$$C_{periph} = \frac{Q_{periph}}{V_{periph}} \tag{45}$$

The following equation computes Q_{total} for mass balance checking:

$$Q_{total} = Q_{central} + Q_{periph} + Q_{elim} \tag{46}$$

Parameters and parameters' scaling

The model parameters have no specific physiological interpretation and are assigned arbitrary default values. Meaningful values should be set by the user. Rate constants are in 1/h, volumes in L.

PBPK model

This model has seven compartments, each with physiological meaning (see Figure 4). Oral dosing is described in the following equations by a first order input from the gut lumen. A version coupling this model to the openCAT model is given in Annex 1.



Figure 4: Schematic representation of the PBPK distribution model.

Main differential equations

Starting with input to the gut, the differential equation governing the drug quantity in the gut lumen (Q_{lumen}) is:

$$\frac{\partial Q_{lumen}}{\partial t} = D_{rate} - K_a \cdot Q_{lumen} \tag{47}$$

where D_{rate} is the oral dosing rate (a model input, in micromoles / hours), and K_a is the absorption rate constant (in 1/hr).

The differential equation for the quantity of drug in the gut tissue (Q_{gut}) is:

$$\frac{\partial Q_{gut}}{\partial t} = F_{gut} \cdot \left(C_{blood} - \frac{C_{gut}}{P_{gut}} \right) + K_a \cdot Q_{lumen} - \frac{V_{max_{gut}} \cdot C_{gut}}{\frac{K_m \cdot P_{gut} \cdot f_{umet_{gut}}}{f_{u_{blood}}} + C_{gut}} \tag{48}$$

where F_{gut} is the blood flow to the gut, P_{gut} is the gut over blood partition coefficient, $V_{max_{gut}}$ is the maximum rate of metabolism in the gut, K_m is Michaelis-Menten coefficient for metabolism.

Correspondingly, the differential for the quantity metabolized in the gut is:

$$\frac{\partial Q_{met_{gut}}}{\partial t} = \frac{V_{max_{gut}} \cdot C_{gut}}{\frac{K_m \cdot P_{gut} \cdot f_{umet_{gut}}}{f_{u_{blood}}} + C_{gut}} \tag{49}$$

The differential equation for the quantity of drug in the liver (Q_{liver}) is:

$$\frac{\partial Q_{liver}}{\partial t} = F_{liver_{art}} \cdot C_{blood} + F_{gut} \cdot \frac{C_{gut}}{P_{gut}} - F_{liver} \cdot \frac{C_{liver}}{P_{liver}} - \frac{V_{max_{liver}} \cdot C_{liver}}{\frac{K_m \cdot P_{liver} \cdot f_{umet_{liver}}}{f_{u_{blood}}} + C_{liver}} \tag{50}$$

where $F_{liver_{art}}$ is the arterial blood flow to the liver, F_{liver} is the sum of $F_{liver_{art}}$ and F_{gut}, P_{liver} is the liver over blood partition coefficient, $V_{max_{liver}}$ is the maximum rate of metabolism in the liver, and $f_{u_{liver}}$ is the fraction unbound in the liver tissue.

The differential for the quantity metabolized in the liver is:

$$\frac{\partial Q_{met_{liver}}}{\partial t} = \frac{V_{max_{liver}} \cdot C_{liver}}{\frac{K_m \cdot P_{liver} \cdot f_{umet_{liver}}}{f_{u_{blood}}} + C_{liver}} \tag{51}$$

The differential equation for the quantity of drug in the kidney (Q_{kidney}) is:

$$\frac{\partial Q_{kidney}}{\partial t} = F_{kidney} \cdot \left(C_{blood} - \frac{C_{kidney}}{P_{kidney}} \right) - K_{e_{kidney}} \cdot f_{u_{blood}} \cdot \frac{C_{kidney}}{P_{kidney}} \tag{52}$$

where F_{kidney} is the blood flow to the kidney, P_{kidney} is the kidney over blood partition coefficient, $K_{e_{kidney}}$ is the excretion flow rate to urine, and $f_{u_{blood}}$ is the fraction unbound in blood (computed as $f_{u_{plasma}}$ divided by the blood over plasma concentration ratio r_{BP}).

The differential for the quantity excreted in urine is:

$$\frac{\partial Q_{urine}}{\partial t} = K_{e_{kidney}} \cdot f_{u_{blood}} \cdot \frac{C_{kidney}}{P_{kidney}} \tag{53}$$

The differential equations for the quantities of drug in the fat (Q_{fat}), poorly perfused tissue (Q_{ppt}), and richly perfused tissue (Q_{rpt}) are respectively:

$$\frac{\partial Q_{adip}}{\partial t} = F_{adip} \cdot \left(C_{blood} - \frac{C_{adip}}{P_{adip}} \right) \tag{54}$$

$$\frac{\partial Q_{ppt}}{\partial t} = F_{ppt} \cdot \left(C_{blood} - \frac{C_{ppt}}{P_{ppt}} \right) \tag{55}$$

$$\frac{\partial Q_{rpt}}{\partial t} = F_{rpt} \cdot \left(C_{blood} - \frac{C_{rpt}}{P_{rpt}} \right) \tag{56}$$

The differential equations for the quantities of drug in blood (Q_{blood}) is:

$$\frac{\partial Q_{blood}}{\partial t} = F_{liver} \cdot \frac{C_{liver}}{P_{liver}} + F_{kidney} \cdot \frac{C_{kidney}}{P_{kidney}} + F_{adip} \cdot \frac{C_{adip}}{P_{adip}} + F_{ppt} \cdot \frac{C_{ppt}}{P_{ppt}} + F_{rpt} \cdot \frac{C_{rpt}}{P_{rpt}} - F_{total} \cdot C_{blood} \tag{57}$$

Miscellaneous equations

Miscellaneous equations are used to keep track of several quantities of interest. The total quantities absorbed by the gut ($Q_{abs_{gut}}$) and reaching the liver *via* the portal vein ($Q_{abs_{portal}}$) are computed respectively by integration of the following differentials:

$$\frac{\partial Q_{abs_{gut}}}{\partial t} = K_a \cdot Q_{lumen} \tag{58}$$

$$\frac{\partial Q_{abs_{portal}}}{\partial t} = F_{gut} \cdot \frac{C_{gut}}{P_{gut}} \quad (59)$$

The integrals (AUCs) of the blood (AUC_{blood}) and reaching the liver *via* the portal vein (AUC_{liver}) are computed respectively as:

$$AUC_{blood} = C_{blood} \quad (60)$$

$$AUC_{liver} = C_{liver} \quad (61)$$

The blood plasma concentration (C_{plasma}) is:

$$C_{plasma} = \frac{C_{blood}}{r_{BP}} \quad (62)$$

where r_{BP} is the blood to plasma concentration ratio.

The instantaneous apparent rate of gut absorption:

$$K_a = \frac{Q_{abs_{portal}}}{Q_{lumen}} \quad (63)$$

The following equations compute Q_{total} for mass balance checking:

$$Q_{elim} = Q_{met_{gut}} + Q_{met_{liver}} + Q_{urine} \quad (64)$$

$$Q_{organs} = Q_{lumen} + Q_{gut} + Q_{liver} + Q_{kidney} + Q_{adip} + Q_{ppt} + Q_{rpt} + Q_{blood} \quad (65)$$

$$Q_{total} = Q_{organs} + Q_{elim} \quad (66)$$

Parameters and parameters' scaling

Many of the parameters used in the above equations are scaled prior to simulation, in order to simplify input and to automatically model correlations between parameters. We give here the corresponding equations. The list of primary physiological parameters (the value of which the user may set to any value for a particular simulation) is

recapitulated in Table 3, together with their units and default values. The drug-specific parameters are listed in Table 4.

Total blood flow (cardiac output, in L/h) is assumed to depend on B_m (the body mass, in kg):

$$F_{total} = sc_{F_{total}} \cdot B_m^{0.75} \tag{67}$$

For compatibility with the openCAT model, the gut blood flow is computed as:

$$F_{gut} = F_{total} \cdot \sum_{i \in O} f_{F_i} \tag{68}$$

where the summation is taken over the set O of gut segments (O = {*stomach*, *duodenum*, *jejunum*, *ileum*, *cecum*, *colon*}). The fraction, f_{F_i}, of cardiac output flowing to tissue segment i is given in Table 3.

The arterial blood flow (L/h) to the liver is given by:

$$F_{liver_{art}} = f_{F_{liver}} \cdot F_{total} \tag{69}$$

The total blood flow (L/h) to the liver is :

$$F_{liver} = F_{liver_{art}} + F_{gut} \tag{70}$$

The blood flows (L/h) to the kidney, fat, and poorly perfused tissue are respectively:

$$F_{kidney} = f_{F_{kidney}} \cdot F_{total} \tag{71}$$

$$F_{adip} = f_{F_{adip}} \cdot F_{total} \tag{72}$$

$$F_{ppt} = f_{F_{ppt}} \cdot F_{total} \tag{73}$$

The blood flow to the well perfused tissue is computed as:

$$F_{wpt} = F_{total} - F_{liver} - F_{kidney} - F_{adip} - F_{ppt} \tag{74}$$

For compatibility with the openCAT model, the gut lumen volume (in L) is computed as:

$$V_{lumen} = B_m \cdot \sum_{i \in O} f_{Bl_i} \tag{75}$$

The fraction of body mass corresponding to each lumen segment i (f_{Bl_i}) is given in Table 3.

Similarly, the gut tissue volume (in L) is computed as:

$$V_{gut} = B_m \cdot \sum_{i \in O} f_{Bw_i} \tag{76}$$

The fraction of body mass corresponding to each gut tissue segment i (f_{Bw_i}) is given in Table 3.

The volumes (L) of the liver, kidney, fat, blood and well perfused tissue are respectively:

$$V_{liver} = f_{Bw_{liver}} \cdot B_m \tag{77}$$

$$V_{kidney} = f_{Bw_{kidney}} \cdot B_m \tag{78}$$

$$V_{adip} = f_{Bw_{adip}} \cdot B_m \tag{79}$$

$$V_{blood} = f_{Bw_{blood}} \cdot B_m \tag{80}$$

$$V_{wpt} = f_{Bw_{wpt}} \cdot B_m \tag{81}$$

The volume of the poorly perfused tissue is computed as:

$$V_{ppt} = B_m - V_{gut} - V_{liver} - V_{kidney} - V_{adip} - V_{wpt} - V_{blood} \tag{82}$$

The Michaelis-Menten maximum rate of metabolism in liver and gut, ($V_{max_{liver}}$ and $V_{max_{gut}}$, in micromol/h) are scaled from *in vitro* values using:

$$V_{max_{liver}} = 60 \times 1000 \cdot V_{max_v} \cdot \mu_{p_{liver}} \cdot V_{liver} \tag{83}$$

$$V_{max_{gut}} = 60 \times 1000 \cdot V_{max_v} \cdot \mu_{p_{gut}} \cdot V_{gut} \tag{84}$$

where V_{max_v} is the *in vitro* measured value (in micromol/min/mg of microsomal proteins), $\mu_{p_{liver}}$ I and $\mu_{p_{gut}}$ are the microsomal abundances in the corresponding organs (in mg/g of tissue).

The *in vivo* Michaelis-Menten constant for metabolism in gut (K_m) is simply the value measured *in vitro*:

$$K_m = K_{m_v} \tag{85}$$

Table 3: Physiological parameters of the PBPK model.

Parameters	Symbols	Units	Default values	References
Body mass	B_m	kg	70	
Cardiac output scaling coefficient	$sc_{F_{total}}$	$L.kg^{-0.75}$	15	
Fractions of cardiac output going to tissues				[1]
liver (arterial)	$f_{F_{liver}}$	-	0.077	
kidney	$f_{F_{kidney}}$	-	0.21	
fat	$f_{F_{adip}}$	-	0.06	
poorly perfused tissues	$f_{F_{ppt}}$	-	0.20	
stomach	$f_{F_{stom}}$	-	0.024	
duodenum	$f_{F_{duod}}$	-	0.016	
jejunum	$f_{F_{jeju}}$	-	0.056	
ileum	$f_{F_{ileum}}$	-	0.033	
cecum	$f_{F_{cecum}}$	-	0.006	
colon	$f_{F_{colon}}$	-	0.038	
Tissue fractions of body mass				[1]
liver	$f_{Bw_{liver}}$	-	0.0243	
kidney	$f_{Bw_{kidney}}$	-	0.0096	
fat	$f_{Bw_{adip}}$	-	0.16	
richly perfused tissues	$f_{Bw_{rpt}}$	-	0.10	
blood	$f_{Bw_{blood}}$	-	0.065	
stomach	$f_{Bw_{stom}}$	-	0.0021	
duodenum	$f_{Bw_{duod}}$	-	0.0003	
jejunum	$f_{Bw_{jeju}}$	-	0.0009	
ileum	$f_{Bw_{ileum}}$	-	0.0006	
cecum	$f_{Bw_{cecum}}$	-	0.0005	
colon	$f_{Bw_{colon}}$	-	0.0048	
Lumina as fractions of body mass				[1]
stomach	$f_{Bl_{stom}}$	-	0.0036	
duodenum	$f_{Bl_{duod}}$	-	0.0003	
jejunum	$f_{Bl_{jeju}}$	-	0.0023	
ileum	$f_{Bl_{ileum}}$	-	0.0032	
cecum	$f_{Bl_{cecum}}$	-	0.0001	
colon	$f_{Bl_{colon}}$	-	0.0051	
Microsomal proteins				
liver	$\mu_{P_{liver}}$	mg / g of tissue	45	[2]
gut	$\mu_{P_{gut}}$	mg / g of tissue	6	[3]

[1]: Perdaems *et al.* Clinical Pharmacokinetics 2010; 49: 239-258.

[2]: Houston *et al.* Toxicology in Vitro 2012; 26: 1265-1271.

[3]: Paine *et al.* The Journal of Pharmacology and Experimental Therapeutics 1997; 283: 1552-1562.

Table 4: Drug-specific parameters of the PBPK model.

Parameters	Symbols	Units	Default values
Molecular mass	M	g/mol	0
Absorption rate constant	K_a	1/hr	0
Absorption lag time	T_{lag}	hr	0
Renal elimination clearance	Ke_{kidney}	L/h	0
Tissue / blood partition coefficients			
gut	P_{gut}	-	1
liver	P_{liver}	-	1
kidney	P_{kidney}	-	1
fat	P_{adip}	-	1
richly perfused tissues	P_{rpt}	-	1
poorly perfused tissues	P_{ppt}	-	1
Plasma over blood concentration ratio	r_{BP}	-	1
Fraction unbound in plasma	$f_{u_{plasma}}$	-	1
Fraction unbound in metabolic systems			
liver	$f_{umet_{liver}}$	-	1
gut	$f_{umet_{gut}}$	-	1
Michaelis-Menten *Vmax*	V_{maxv}	µmol/min/mg microsomal prot	0
Michaelis-Menten *Km*	K_{mv}	microM	0

Annex 1 – GNU MCSim codes

Two-compartment, first order absorption model

```
# ================================================================================
# Two-compartment model with first-order absorption rate and
# linear elimination (metabolism or other route of elimination)
#
# version 1
#
# Units:
# - time in hours
# - volumes in liters
# - masses of substances in micromoles
# - concentrations of substances in microM
#
# Frederic Bois - Dec 2016
# ================================================================================

States  = {Q_central,       # Quantity in central compartment (micromoles)
           Q_periph,        # ~           peripheral compartment
           Q_elim};         # ~        eliminated

Outputs = {C_central,       # Concentration in central compartment (microM)
           C_periph,        # ~                peripheral compartment
           Q_total};        # Total quantity for mass balance

Inputs  = {Oral_input};     # Drug input in micromoles


# ------------------------
# Parameters
# ------------------------

# Oral input modeling

IngDose    = 0.0; # ingested input (micromoles)
Period     = 0.0; # period of the exposure/no exposure cycle (h)
Tlag       = 0.0; # Absorption lagtime (h)
Ka         = 0.0; # Intestinal absorption rate constant (1/h)
Oral_input = PerExp (IngDose, Period, Tlag, Ka);

# Elimination rate constant (1/h)
Ke = 0;
```

```
# Volumes (L)
V_central = 1;
V_periph  = 1;

# Transfer rate constants between compartments (1/h)
Kc2p = 0;
Kp2c = 0;

Dynamics {
  dt (Q_elim)    = Ke * Q_central;
  dt (Q_periph)  = Kc2p * Q_central - Kp2c * Q_periph;
  dt (Q_central) = Ka * Oral_input - dt(Q_elim) - dt(Q_periph);
}

CalcOutputs {
  C_central = Q_central / V_central;
  C_periph  = Q_periph  / V_periph;
  Q_total   = Q_central + Q_periph + Q_elim;
}

End.
```

PBPK, first order absorption model

```
# ================================================================================
# Six-compartment PBPK model with first-order absorption from the gut.
# No difference between arterial and venous blood, no inhalation.
# Transport to tissues is flow limited.
#
# version 1.
#
# Units:
# - time in hours
# - volumes in liters
# - masses of substances in micromoles
# - concentrations of substances in microM
#
# Frederic Bois - Dec 2016
# ================================================================================

States  = {Q_kid,          # Quantity in kidney (micromoles)
           Q_rpt,          # ~        richly perfused tissues (viscera)
           Q_liver,        # ~        liver
           Q_adip,         # ~        adipose tissue
           Q_ppt,          # ~        poorly perfused tissues (muscle and skin)
           Q_blood,        # ~        blood
           Q_gut_lu,       # ~        gut  lumen
           Q_gut,          # ~        gut
           Q_absorb_gut,   # ~        absorbed from gut lumen
           Q_absorb_pv,    # ~        reaching the portal vein
           Q_elim_kid,     # ~        urine excreted or metabolized in kidney
           Q_met_liver,    # ~        metabolized in liver
           Q_met_gut,      # ~        metabolized in gut
           AUC_blood,      # Integral of blood concentration
           AUC_liver};     # ~           liver ~

Outputs = {C_plasma,       # Concentration in blood plasma (microM)
           C_kid,          # ~                kidney
           C_rpt,          # etc.
           C_liver,
           C_adip,
           C_ppt,
           C_blood,
           C_gut_lu,
           C_gut,
           Q_elim,         # Quantity metabolized or excreted
           Q_organ,        # ~ in the body (including lumina)
           Q_total,        # ~ in the body and metabolized or excreted
           Instant_Ka_gi}; # Instantaenous net absorption ratio by g.i. tract
```

```
Inputs  = {Oral_dose_rate};# in micromoles / hours


# ------------------------
# Physiological parameters
# ------------------------

# Body mass (kg)
BDM = 70;

# Total blood flow scaling coefficient to body mass
sc_F_total = 15;

# Fractions of total blood flow going to tissues
f_Flow_adip      = 0.06;
f_Flow_kid       = 0.21;
f_Flow_ppt       = 0.20;
f_Flow_liver_art = 0.077;
f_Flow_stom      = 0.024;
f_Flow_duod      = 0.016;
f_Flow_jeju      = 0.056;
f_Flow_ileon     = 0.033;
f_Flow_cecum     = 0.006;
f_Flow_colon     = 0.038;

# Tissue fractions of body mass
f_BDM_adip  = 0.16;
f_BDM_blood = 0.065;
f_BDM_kid   = 0.0096;
f_BDM_rpt   = 0.1;
f_BDM_liver = 0.0243;
f_BDM_stom  = 0.0021;
f_BDM_duod  = 0.0003;
f_BDM_jeju  = 0.0009;
f_BDM_ileon = 0.0006;
f_BDM_cecum = 0.0005;
f_BDM_colon = 0.0048;

# Lumina as fractions of body mass
f_BDM_stom_lu  = 0.0036;
f_BDM_duod_lu  = 0.0003;
f_BDM_jeju_lu  = 0.0023;
f_BDM_ileon_lu = 0.0032;
f_BDM_cecum_lu = 0.0001;
f_BDM_colon_lu = 0.0051;

# Microsomal proteins (mg / g of tissue)
```

```
MicroProt_liver  = 45; # Houston 2012
MicroProt_gut    = 6;  # Paine 1997


#-------------------------------------
# Susbstance-specific parameters
#-------------------------------------

# Molecular mass (g/mol)
MM;

# Intestinal absortion rate (1/h)
Kabs_gut;

# Administration lagtime, if needed
Tlag;

# Tissue / blood partition coefficients
PC_adip;
PC_rpt;
PC_ppt;
PC_kid;
PC_liver;
PC_gut;

# Plasma / blood concentration ratio
Ratio_BP = 1;

# Fractions unbound
Fu_plasma = 1.0;

Fu_met_liver = 1.0; # to liver in vitro metabolic system (e.g. microsomes)
Fu_met_gut   = 1.0; # to gut   ~

# Michaelis-Menten Vmax (micromol/min/mg microsomal proteins)
Vmax_vitro;

# Michaelis-Menten Km (microM)
Km_vitro;

# Renal elimination clearance (L/h)
Ke_kid;


#-------------------------------------
# Scaled parameters
#-------------------------------------
```

```
# Blood flows
F_total;

# Volumes
V_adip;
V_blood;
V_kid;
V_liver;
V_rpt;
V_ppt;
V_gut;

# Luminal volumes in gi tract
V_gut_lu;

# Blood flows
Flow_adip;
Flow_kid;
Flow_liver;
Flow_liver_art;
Flow_rpt;
Flow_ppt;
Flow_gut;

# Fraction unbound in blood
Fu_blood;

# In vivo liver and gi tract Vmax
Vmax_vivo_liver;
Vmax_vivo_gut;

# In vivo Km
Km_vivo;


Initialize {

  # Total blood flow
  F_total = sc_F_total * pow(BDM, 0.75);

  # Volumes
  V_adip     = BDM * f_BDM_adip;
  V_blood    = BDM * f_BDM_blood;
  V_kid      = BDM * f_BDM_kid;
  V_liver    = BDM * f_BDM_liver;
  V_rpt      = BDM * f_BDM_rpt;
  V_gut      = BDM * (f_BDM_stom  + f_BDM_duod + f_BDM_jeju + f_BDM_ileon +
                      f_BDM_cecum + f_BDM_colon);
```

```
  V_ppt = BDM - V_adip  - V_blood - V_kid - V_liver - V_rpt - V_gut;

  # Luminal volume
  V_gut_lu  = BDM * (f_BDM_stom_lu  + f_BDM_duod_lu  + f_BDM_jeju_lu +
                     f_BDM_ileon_lu + f_BDM_cecum_lu + f_BDM_colon_lu);

  # Blood flows
  Flow_adip       = F_total * f_Flow_adip;
  Flow_kid        = F_total * f_Flow_kid;
  Flow_ppt        = F_total * f_Flow_ppt;
  Flow_liver_art  = F_total * f_Flow_liver_art;
  Flow_gut        = F_total * (f_Flow_stom  + f_Flow_duod  + f_Flow_jeju +
                               f_Flow_ileon + f_Flow_cecum + f_Flow_colon);

  Flow_liver      = Flow_liver_art + Flow_gut;

  Flow_rpt = F_total - Flow_adip - Flow_kid - Flow_ppt - Flow_liver;

  # Fraction unbound in blood
  Fu_blood = Fu_plasma / Ratio_BP;

  # Metabolic parameters, scaled from in vitro values
  Vmax_vivo_liver = Vmax_vitro * 60 * MicroProt_liver * 1000 * V_liver;
  Vmax_vivo_gut   = Vmax_vitro * 60 * MicroProt_gut   * 1000 * V_gut;

  Km_vivo = Km_vitro;

} # End of model scaling


Dynamics {

  # Concentrations in compartments
  C_adip  = Q_adip  / V_adip;
  C_kid   = Q_kid   / V_kid;
  C_rpt   = Q_rpt   / V_rpt;
  C_ppt   = Q_ppt   / V_ppt;
  C_liver = Q_liver / V_liver;
  C_gut   = Q_gut   / V_gut;

  # Blood concentration
  C_blood = Q_blood / V_blood;

  # Blood concentration at compartment exit
  Ctb_kid   = C_kid   / PC_kid;
  Ctb_liver = C_liver / PC_liver;
  Ctb_rpt   = C_rpt   / PC_rpt;
```

```
Ctb_ppt   = C_ppt   / PC_ppt;
Ctb_adip  = C_adip  / PC_adip;
Ctb_gut   = C_gut   / PC_gut;

# Concentrations in gut lumen
C_gut_lu  = Q_gut_lu  / V_gut_lu;

# Transfer from gut lumen to gut tissue
Rate_in = Kabs_gut  * Q_gut_lu;

# Transfers from tissues to portal vein
Rate_in_portvein = Flow_gut * Ctb_gut;

# Differential equations

# Quantities absorbed
dt (Q_absorb_gut) = Rate_in;

# Quantity reaching the portal vein
dt (Q_absorb_pv) = Rate_in_portvein;

# Elimination and metabolism

dt (Q_elim_kid)  = Ke_kid * Ctb_kid * Fu_blood;

dt (Q_met_liver) = Vmax_vivo_liver * Ctb_liver /
                   (Km_vivo * Fu_met_liver / Fu_blood + Ctb_liver);

dt (Q_met_gut)   = Vmax_vivo_gut * Ctb_gut /
                   (Km_vivo * Fu_met_gut / Fu_blood + Ctb_gut);

# Distribution

dt (Q_kid)    = Flow_kid  * (C_blood - Ctb_kid) - dt (Q_elim_kid);
dt (Q_adip)   = Flow_adip * (C_blood - Ctb_adip);
dt (Q_ppt)    = Flow_ppt  * (C_blood - Ctb_ppt);
dt (Q_rpt)    = Flow_rpt  * (C_blood - Ctb_rpt);
dt (Q_liver)  = Flow_liver_art * C_blood + Flow_gut * Ctb_gut -
                Flow_liver * Ctb_liver - dt (Q_met_liver);

dt(Q_blood)   = Flow_rpt  * Ctb_rpt  + Flow_ppt   * Ctb_ppt +
                Flow_adip * Ctb_adip + Flow_liver * Ctb_liver +
                Flow_kid  * Ctb_kid  - F_total * C_blood;

dt(Q_gut_lu) = Oral_dose_rate - Rate_in;

dt(Q_gut) = Flow_gut * (C_blood - Ctb_gut) + Rate_in - dt (Q_met_gut);
```

```
  # AUCs

  dt(AUC_blood) = C_blood;
  dt(AUC_liver) = C_liver;

}

CalcOutputs {

  C_plasma = (C_blood > 0 ? C_blood / Ratio_BP : 1E-10);

  # Mass balance checking

  Q_elim  = Q_elim_kid + Q_met_liver + Q_met_gut;

  Q_organ = Q_kid   + Q_rpt    + Q_liver + Q_adip + Q_ppt +
            Q_blood + Q_gut_lu + Q_gut;

  Q_total = Q_organ + Q_elim;

  Instant_Ka_gi = (Q_gut_lu > 0 ? Q_absorb_pv / Q_gut_lu : 0);

}
End.
```

Two-compartment, openCAT absorption model (v3)

```
# ============================================================================
# Two-compartment PBPK model with openCAT gut model.
# Linear metabolism
#
# Effective permeability is assumed to be the same for each segment
# of the gi tract
#
# version 2: drug release and dissolution equations added
# version 3: adding the liver between gut and central (see Huang 2009)
#            adding apical influx transport in gut compartments and liver
#            adding apical efflux ~
#            adding fractions unbound in gut and liver compartments
#
# Units:
# - time in hours
# - volumes in liters
# - masses of substances in micromoles
# - concentrations of substances in microM
#
# Frederic Bois - 2017
# ============================================================================

States  = {Q_central,      # Quantity in central compartment (micromoles)
           Q_periph,       # ~           peripheral compartment
         Q_liver,        # ~           liver
           # Q_to_release, # ~        to release in stomach
           Q_released,     # ~        released in stomach
           Q_stom_lu_d,    # ~        in stomach  lumen (dissolved)
           Q_duod_lu_d,    # ~           duodenum ~     ~
           Q_jeju_lu_d,    # ~           jejunum  ~     ~
           Q_ileum_lu_d,   # ~           ileum    ~     ~
           Q_cecum_lu_d,   # ~           cecum    ~     ~
           Q_colon_lu_d,   # ~           colon    ~     ~
           Q_stom_lu_u,    # ~           stomach  lumen (undissolved)
           Q_duod_lu_u,    # ~           duodenum ~
           Q_jeju_lu_u,    # ~           jejunum  ~
           Q_ileum_lu_u,   # ~           ileum    ~
           Q_cecum_lu_u,   # ~           cecum    ~
           Q_colon_lu_u,   # ~           colon    ~
           Q_stom_ep,      # ~           stomach epithelium
           Q_duod_ep,
           Q_jeju_ep,
           Q_ileum_ep,
           Q_cecum_ep,
           Q_colon_ep,
```

```
            Q_stom_w,       # ~           stomach wall
            Q_duod_w,
            Q_jeju_w,
            Q_ileum_w,
            Q_cecum_w,
            Q_colon_w,
            Q_absorb_stom,  # ~           absorbed from stomach lumen
            Q_absorb_duod,
            Q_absorb_jeju,
            Q_absorb_ileum,
            Q_absorb_cecum,
            Q_absorb_colon,
            Q_absorb_pv,    # ~           reaching the portal vein
            Q_elim_central, # ~           metabolized or excreted from central
            Q_met_stom,
            Q_met_duod,
            Q_met_jeju,
            Q_met_ileum,
            Q_met_cecum,
            Q_met_colon,
            Q_met_liver,
            Q_feces_d,      # ~           excreted dissolved   in feces
            Q_feces_u};     # ~           excreted undissolved in feces

Outputs = {C_central,       # Concentration in central compartment (microM)
            C_periph,       #                   peripheral
            C_stom_lu_d,    #                   stomach lumen (dissolved)
            C_duod_lu_d,
            C_jeju_lu_d,
            C_ileum_lu_d,
            C_cecum_lu_d,
            C_colon_lu_d,
            C_stom_ep,
            C_duod_ep,
            C_jeju_ep,
            C_ileum_ep,
            C_cecum_ep,
            C_colon_ep,
            C_stom_w,
            C_duod_w,
            C_jeju_w,
            C_ileum_w,
            C_cecum_w,
            C_colon_w,
            Q_gi_lu_d,      # Quantity in g.i. tract lumen (dissolved)
            Q_gi_lu_u,      # ~                           (undissolved)
            Q_gi_ep,        # ~                     epithelium
            Q_gi_w,         # ~                     wall
```

```
           Q_absorb,       # ~ absorbed by the g.i. tract
           Q_elim,         # ~ metabolized or excreted
           Q_met_gi,       # ~ metabolized in the g.i. tract
           Q_elim_gi,      # ~ metabolized in or excreted from the g.i. tract
           Q_organ,        # ~ in the body (including lumina)
           Q_total,        # ~ in the body and metabolized or excreted
           Release_rate,   # Instantaneous release rate for delayed forms
           Instant_Ka_gi}; # Instantaenous net absorption ratio by g.i. tract

Inputs  = {Oral_dose_rate, IV_dose_rate, # immediate dose, in micromoles / h
           PO_dose};                     # delayed oral dose, in micromoles

# ------------------------
# Physiological parameters
# ------------------------

# Body mass (kg)
BDM = 70;

# Total blood flow scaling coefficient to body mass
sc_F_total = 15;

# Fractions of total blood flow going to tissues (ref Perdaems 2010)
f_Flow_stom  = 0.024;
f_Flow_duod  = 0.016;
f_Flow_jeju  = 0.056;
f_Flow_ileum = 0.033;
f_Flow_cecum = 0.006;
f_Flow_colon = 0.038;
f_Flow_liver = 0.250;

# Tissue fractions of body mass (ref Perdaems 2010)
f_BDM_stom  = 0.0021;
f_BDM_duod  = 0.0003;
f_BDM_jeju  = 0.0009;
f_BDM_ileum = 0.0006;
f_BDM_cecum = 0.0005;
f_BDM_colon = 0.0048;
f_BDM_liver = 0.0243;

# Lumina as fractions of body mass (ref Perdaems 2010)
f_BDM_stom_lu  = 0.0036;
f_BDM_duod_lu  = 0.0003;
f_BDM_jeju_lu  = 0.0023;
f_BDM_ileum_lu = 0.0032;
f_BDM_cecum_lu = 0.0001;
f_BDM_colon_lu = 0.0051;
```

```
# Lengths of gi tract segments (decimeters, dm)
Length_stom  = 2.83;  # Ando 2015
Length_duod  = 1.41;  # Ando 2015
Length_jeju  = 11.68; # Ando 2015
Length_ileum = 17.52; # Ando 2015
Length_cecum = 1.7;   # Perdaems 2010
Length_colon = 11.0;  # ICRP 2002

# Radii of gi tract segments (dm)
Radius_stom  = 0.967; # Ando 2015
Radius_duod  = 0.153; # Ando 2015
Radius_jeju  = 0.137; # mean of the two Ando values
Radius_ileum = 0.098; # mean of the three Ando values
Radius_cecum = 0.35;  # Perdaems 2010
Radius_colon = 0.25;  # Perdaems 2010
# Transit half-lives in lumina (ref Perdaems 2010) (hours)
T12_stom_lu  = 0.25;
T12_duod_lu  = 0.25;
T12_jeju_lu  = 1.02;
T12_ileum_lu = 2.04;
T12_cecum_lu = 4.55;
T12_colon_lu = 13.5;

# pH of luminal contents (ref Perdaems 2010)
pH_stom  = 1.7;
pH_duod  = 6;
pH_jeju  = 6.5;
pH_ileum = 7.4;
pH_cecum = 5.9;
pH_colon = 7;

# Microsomal proteins (mg / g of tissue)
MicroProt_stom    = 0;
MicroProt_duod    = 18;   # Paine 1997
MicroProt_jeju    = 25;   # Paine 1997
MicroProt_ileum   = 24;   # Paine 1997
MicroProt_cecum   = 0.0;
MicroProt_colon   = 0.0;
MicroProt_liver   = 45;   # Houston 2012

# G.i. tract epithelial thickness (dm)
H_ep = 5.25E-4;

#-----------------------------------------------
# Substance- and formulation specific parameters
#-----------------------------------------------

# Molecular mass (g/mol)
```

```
MM;

# Acido-basic type: 0 = neutral, 1 = base, 2 = acid, 3 = ampholyte
AB_type; # neutral is the default

# Ionization constants
pKb; # basic equilibrium constant
pKa; # acid  equilibrium constant

# Solute molal volume at normal boiling point (ml/mole) (about twice MM...)
Mol_vol; # no default value

# Dosage forms: 0/1 switches, mutually exclusive,
# only one on them should be set to 1
G_immediate_d = 1; # immediate release, dissolved (default)
G_immediate_u;     # immediate release, undissolved
G_delayed_d;       # delayed release,   dissolved
G_delayed_u;       # delayed release,   undissolved

# Administration lagtime, if needed (to use in an input function)
Tlag;

# Weibull delayed release parameters
Weibull_slope = 1; # avoid division by zero
Weibull_scale = 1; # avoid division by zero
Weibull_lag;

# First order delayed release rate constant
# K_release; # unused, the equation is commented out

# Galenic radius (microm)
G_Radius  = 25.0;

# Powder density (g/ml)
G_Density = 1.2;

# Intrinsic water solubility (saturation concentration) (microg/L)
Solubility;

# Precipitation rate (1/h)
K_precip;

# Absorption on(1)/off(0) switches
f_Abs_stom;
f_Abs_duod;
f_Abs_jeju;
f_Abs_ileum;
f_Abs_cecum;
```

```
f_Abs_colon;

# Effective permeability of g.i. tract epithelia
Peff;

# Excretion over absorption rate constant ratios
Ke_over_a_epit; # between lumen and epithelium
Ke_over_a_tiss; # between epithelium and underlying wall tissue

# GI tract apical efflux Vmax
Vmax_eff_stom  = 0;
Vmax_eff_duod  = 0;
Vmax_eff_jeju  = 0;
Vmax_eff_ileum = 0;
Vmax_eff_cecum = 0;
Vmax_eff_colon = 0;
Vmax_eff_liver = 0;

# GI tract apical influx Vmax
Vmax_inf_stom  = 0;
Vmax_inf_duod  = 0;
Vmax_inf_jeju  = 0;
Vmax_inf_ileum = 0;
Vmax_inf_cecum = 0;
Vmax_inf_colon = 0;
Vmax_inf_liver = 0;

# GI tract apical efflux Km
Km_eff = 1;

# GI tract apical influx Km
Km_inf = 1;

# Plasma / blood concentration ratio
Ratio_BP = 1;

# Fractions unbound
Fu_plasma = 1;    # fraction unbound in plasma
Fu_stom   = 1;    # etc.
Fu_duod   = 1;
Fu_jeju   = 1;
Fu_ileum  = 1;
Fu_cecum  = 1;
Fu_colon  = 1;
Fu_liver  = 1;

Fu_vitro_stom  = 1.0; # in stomach in vitro metabolic system (e.g. microsomes)
Fu_vitro_duod  = 1.0; # etc.
```

```
Fu_vitro_jeju  = 1.0;
Fu_vitro_ileum = 1.0;
Fu_vitro_cecum = 1.0;
Fu_vitro_colon = 1.0;
Fu_vitro_liver = 1.0;

# Unbound tissue / unbound blood partition coefficients
Kpuu_stom  = 1;
Kpuu_duod  = 1;
Kpuu_jeju  = 1;
Kpuu_ileum = 1;
Kpuu_cecum = 1;
Kpuu_colon = 1;
Kpuu_liver = 1;

# Volumes (L)
V_central = 1.0;
V_periph  = 1.0;

# Transfer rate constants between central and peripheral compartments (1/h)
Kc2p;
Kp2c;

# Metabolism Michaelis-Menten Vmax (micromol/min/mg microsomal proteins)
Vmax_met_vitro;

# Metabolism Michaelis-Menten Km (microM)
Km_met_vitro;

# Elimination rate constant from central (1/h)
Kelim;

#-------------------------------------------------------------------------
# Scaled physiological G.I. tract parameters, computed in Initialize, below
#-------------------------------------------------------------------------

# Blood flows
F_total;

# Volumes (L)
V_stom;
V_duod;
V_jeju;
V_ileum;
V_cecum;
V_colon;
V_liver;
```

```
# Tissue volumes in gi tract
V_stom_w;
V_duod_w;
V_jeju_w;
V_ileum_w;
V_cecum_w;
V_colon_w;

# Luminal volumes in gi tract
V_stom_lu;
V_colon_lu;
V_duod_lu;
V_jeju_lu;
V_ileum_lu;
V_cecum_lu;

# Epithelial volumes in gi tract
V_stom_ep;
V_duod_ep;
V_jeju_ep;
V_ileum_ep;
V_cecum_ep;
V_colon_ep;

# Epithelial surface area in gi tract (dm^2)
SA_stom;
SA_duod;
SA_jeju;
SA_ileum;
SA_cecum;
SA_colon;

# Intestinal transit rates
Kt_stom;
Kt_duod;
Kt_jeju;
Kt_ileum;
Kt_cecum;
Kt_colon;

# Blood flows
Flow_portvein;
Flow_art_liv;
Flow_stom;
Flow_duod;
Flow_jeju;
Flow_ileum;
Flow_cecum;
```

```
Flow_colon;

# GI tract absorption flows
Ka_stom;
Ka_duod;
Ka_jeju;
Ka_ileum;
Ka_cecum;
Ka_colon;

# Fraction unbound in blood
Fu_blood;

# In vivo GI tract and liver Vmax for metabolism
Vmax_met_stom;
Vmax_met_duod;
Vmax_met_jeju;
Vmax_met_ileum;
Vmax_met_cecum;
Vmax_met_colon;
Vmax_met_liver;

# In vivo Km for metabolism
Km_met;

# Diffusion layer thickness
Diff_thickness;

# Diffusion coefficient at infinite dilution
Diffusivity;

# Baseline dissolution rate constant (1/microM/hr)
K_diss;

# Saturation concentrations
Csat_stom;
Csat_duod;
Csat_jeju;
Csat_ileum;
Csat_cecum;
Csat_colon;

Initialize {

  # Total blood flow
  F_total = sc_F_total * pow(BDM, 0.75);

  V_liver = BDM * f_BDM_liver;
```

```
# Volumes of gi tract tissues
V_stom_w   = BDM * f_BDM_stom;
V_duod_w   = BDM * f_BDM_duod;
V_jeju_w   = BDM * f_BDM_jeju;
V_ileum_w  = BDM * f_BDM_ileum;
V_cecum_w  = BDM * f_BDM_cecum;
V_colon_w  = BDM * f_BDM_colon;

# Blood flows
Flow_stom     = F_total * f_Flow_stom;
Flow_duod     = F_total * f_Flow_duod;
Flow_jeju     = F_total * f_Flow_jeju;
Flow_ileum    = F_total * f_Flow_ileum;
Flow_cecum    = F_total * f_Flow_cecum;
Flow_colon    = F_total * f_Flow_colon;
Flow_portvein = Flow_stom  + Flow_duod  + Flow_jeju +
                Flow_ileum + Flow_cecum + Flow_colon;
Flow_art_liv  = F_total * f_Flow_liver - Flow_portvein;

# Epithelial surface areas
SA_stom   = Length_stom  * 2 * 3.1416 * Radius_stom;
SA_duod   = Length_duod  * 2 * 3.1416 * Radius_duod;
SA_jeju   = Length_jeju  * 2 * 3.1416 * Radius_jeju;
SA_ileum  = Length_ileum * 2 * 3.1416 * Radius_ileum;
SA_cecum  = Length_cecum * 2 * 3.1416 * Radius_cecum;
SA_colon  = Length_colon * 2 * 3.1416 * Radius_colon;

# Luminal volumes
V_stom_lu  = f_BDM_stom_lu  * BDM;
V_duod_lu  = f_BDM_duod_lu  * BDM;
V_jeju_lu  = f_BDM_jeju_lu  * BDM;
V_ileum_lu = f_BDM_ileum_lu * BDM;
V_cecum_lu = f_BDM_cecum_lu * BDM;
V_colon_lu = f_BDM_colon_lu * BDM;

# Intestinal transit rates
Kt_stom  = (log(2.0) * V_stom_lu  / T12_stom_lu);
Kt_duod  = (log(2.0) * V_duod_lu  / T12_duod_lu);
Kt_jeju  = (log(2.0) * V_jeju_lu  / T12_jeju_lu);
Kt_ileum = (log(2.0) * V_ileum_lu / T12_ileum_lu);
Kt_cecum = (log(2.0) * V_cecum_lu / T12_cecum_lu);
Kt_colon = (log(2.0) * V_colon_lu / T12_colon_lu);

# Epithelial volumes
V_stom_ep     = H_ep * SA_stom;
V_duod_ep     = H_ep * SA_duod;
V_jeju_ep     = H_ep * SA_jeju;
```

```
V_ileum_ep    = H_ep * SA_ileum;
V_cecum_ep    = H_ep * SA_cecum;
V_colon_ep    = H_ep * SA_colon;

# Fraction unbound in blood
Fu_blood = Fu_plasma / Ratio_BP;

# Metabolic parameters, scaled from in vitro values
Vmax_met_stom  = Vmax_met_vitro * 60 * MicroProt_stom  * V_stom_ep  * 1000;
Vmax_met_duod  = Vmax_met_vitro * 60 * MicroProt_duod  * V_duod_ep  * 1000;
Vmax_met_jeju  = Vmax_met_vitro * 60 * MicroProt_jeju  * V_jeju_ep  * 1000;
Vmax_met_ileum = Vmax_met_vitro * 60 * MicroProt_ileum * V_ileum_ep * 1000;
Vmax_met_cecum = Vmax_met_vitro * 60 * MicroProt_cecum * V_cecum_ep * 1000;
Vmax_met_colon = Vmax_met_vitro * 60 * MicroProt_colon * V_colon_ep * 1000;
Vmax_met_liver = Vmax_met_vitro * 60 * MicroProt_liver * V_liver    * 1000;

Km_met = Km_met_vitro;

# Diffusion layer thickness (bounded between 5 and 30 microns)
# See Arav, Drug Development and Industrial Pharmacy, 2012; 38:940-951
Diff_thickness = (G_Radius < 5 ? 5 : G_Radius > 30 ? 30 : G_Radius);

# Diffusion coefficient at infinite dilution (cm2/sec) at 36 degrees Celsius
# in water; 1.09E-4 = 7.4E-8 * 309.15 * sqrt(18) / 0.89; 309.15 is 36 degrees
# Celsius in Kelvin; 18 it the molecular weight of water; 0.89 is water
# viscosity in centopoises. Ref: Wilke, 1955
Diffusivity = 1.09E-4 * pow(Mol_vol, -0.6);

# Baseline dissolution rate constant (1/microM/hr)
K_diss = 1080 * MM * Diffusivity / (G_Density * G_Radius * Diff_thickness);

# Saturation concentrations (according to Henderson-Hasselbach)
# note: AB_type: 0 = neutral, 1 = base, 2 = acid, 3 = ampholyte
Csat_stom  = Solubility * (
 AB_type == 1 ? 1 + pow(10, pKb - pH_stom) :

 AB_type == 2 ? 1 + pow(10, pH_stom - pKa) :

 AB_type == 3 ? 1 + pow(10, pKb - pH_stom) + pow(10, pH_stom - pKa) : 1);

Csat_duod   =  Solubility * (
 AB_type == 1 ? 1 + pow(10, pKb - pH_duod) :

 AB_type == 2 ? 1 + pow(10, pH_duod - pKa) :

 AB_type == 3 ? 1 + pow(10, pKb - pH_duod) + pow(10, pH_duod - pKa) : 1);

Csat_jeju   =  Solubility * (
```

```
   AB_type == 1 ? 1 + pow(10, pKb - pH_jeju) :

   AB_type == 2 ? 1 + pow(10, pH_jeju - pKa) :

   AB_type == 3 ? 1 + pow(10, pKb - pH_jeju) + pow(10, pH_jeju - pKa) : 1);

  Csat_ileum  =  Solubility * (
   AB_type == 1 ? 1 + pow(10, pKb - pH_ileum) :

   AB_type == 2 ? 1 + pow(10, pH_ileum - pKa) :

   AB_type == 3 ? 1 + pow(10, pKb - pH_ileum) + pow(10, pH_ileum - pKa) : 1);

  Csat_cecum  =  Solubility * (
   AB_type == 1 ? 1 + pow(10, pKb - pH_cecum) :

   AB_type == 2 ? 1 + pow(10, pH_cecum - pKa) :

   AB_type == 3 ? 1 + pow(10, pKb - pH_cecum) + pow(10, pH_cecum - pKa) : 1);

  Csat_colon  =  Solubility * (
   AB_type == 1 ? 1 + pow(10, pKb - pH_colon) :

   AB_type == 2 ? 1 + pow(10, pH_colon - pKa) :

   AB_type == 3 ? 1 + pow(10, pKb - pH_colon) + pow(10, pH_colon - pKa) : 1);

  # GI tract absorption flows
  Ka_stom  = Peff * SA_stom  * f_Abs_stom;
  Ka_duod  = Peff * SA_duod  * f_Abs_duod;
  Ka_jeju  = Peff * SA_jeju  * f_Abs_jeju;
  Ka_ileum = Peff * SA_ileum * f_Abs_ileum;
  Ka_cecum = Peff * SA_cecum * f_Abs_cecum;
  Ka_colon = Peff * SA_colon * f_Abs_colon;

} # End of model scaling

Dynamics {

  # Delayed release modeling

  # Weibull delayed release (see TNO intestinal models). This should be
  # doable with a PerExp input...
  Release_rate = PO_dose * Weibull_slope / Weibull_scale *
                 pow((t - Weibull_lag) / Weibull_scale, Weibull_slope - 1) *
                 exp(-pow((t - Weibull_lag) / Weibull_scale, Weibull_slope));

  # First order delayed release, off
```

```
# Release_rate = K_release * Q_to_release;

# Concentrations dissolved in gi tract lumina
C_stom_lu_d  = Q_stom_lu_d  / V_stom_lu;
C_duod_lu_d  = Q_duod_lu_d  / V_duod_lu;
C_jeju_lu_d  = Q_jeju_lu_d  / V_jeju_lu;
C_ileum_lu_d = Q_ileum_lu_d / V_ileum_lu;
C_cecum_lu_d = Q_cecum_lu_d / V_cecum_lu;
C_colon_lu_d = Q_colon_lu_d / V_colon_lu;

# Concentrations undissolved in gi tract lumina
C_stom_lu_u  = Q_stom_lu_u  / V_stom_lu;
C_duod_lu_u  = Q_duod_lu_u  / V_duod_lu;
C_jeju_lu_u  = Q_jeju_lu_u  / V_jeju_lu;
C_ileum_lu_u = Q_ileum_lu_u / V_ileum_lu;
C_cecum_lu_u = Q_cecum_lu_u / V_cecum_lu;
C_colon_lu_u = Q_colon_lu_u / V_colon_lu;

# Concentrations in gi tract epithelia
C_stom_ep  = Q_stom_ep  / V_stom_ep;
C_duod_ep  = Q_duod_ep  / V_duod_ep;
C_jeju_ep  = Q_jeju_ep  / V_jeju_ep;
C_ileum_ep = Q_ileum_ep / V_ileum_ep;
C_cecum_ep = Q_cecum_ep / V_cecum_ep;
C_colon_ep = Q_colon_ep / V_colon_ep;

# Concentrations unbound in gi tract epithelia
Cu_stom_ep  = C_stom_ep  * Fu_stom;
Cu_duod_ep  = C_duod_ep  * Fu_duod;
Cu_jeju_ep  = C_jeju_ep  * Fu_jeju;
Cu_ileum_ep = C_ileum_ep * Fu_ileum;
Cu_cecum_ep = C_cecum_ep * Fu_cecum;
Cu_colon_ep = C_colon_ep * Fu_colon;

# Concentrations in gi tract walls
C_stom_w  = Q_stom_w  / V_stom_w;
C_duod_w  = Q_duod_w  / V_duod_w;
C_jeju_w  = Q_jeju_w  / V_jeju_w;
C_ileum_w = Q_ileum_w / V_ileum_w;
C_cecum_w = Q_cecum_w / V_cecum_w;
C_colon_w = Q_colon_w / V_colon_w;
C_liver   = Q_liver   / V_liver;

# Concentrations unbound in gi tract walls
Cu_stom_w  = C_stom_w  * Fu_stom;
Cu_duod_w  = C_duod_w  * Fu_duod;
Cu_jeju_w  = C_jeju_w  * Fu_jeju;
Cu_ileum_w = C_ileum_w * Fu_ileum;
```

```
Cu_cecum_w = C_cecum_w * Fu_cecum;
Cu_colon_w = C_colon_w * Fu_colon;

# Concentration unbound in liver
Cu_liver   = C_liver   * Fu_liver;

# Concentrations unbound in blood at gi tract segments exit
Cu_stom_b  = Cu_stom_w  / Kpuu_stom;
Cu_duod_b  = Cu_duod_w  / Kpuu_duod;
Cu_jeju_b  = Cu_jeju_w  / Kpuu_jeju;
Cu_ileum_b = Cu_ileum_w / Kpuu_ileum;
Cu_cecum_b = Cu_cecum_w / Kpuu_cecum;
Cu_colon_b = Cu_colon_w / Kpuu_colon;

# Concentration unbound in blood at liver exit
Cu_liver_b = Cu_liver   / Kpuu_liver;

# Concentrations in blood at gi tract segments exit
C_stom_b  = Cu_stom_b  / Fu_blood;
C_duod_b  = Cu_duod_b  / Fu_blood;
C_jeju_b  = Cu_jeju_b  / Fu_blood;
C_ileum_b = Cu_ileum_b / Fu_blood;
C_cecum_b = Cu_cecum_b / Fu_blood;
C_colon_b = Cu_colon_b / Fu_blood;

# Concentration in blood at liver exit
C_liver_b = Cu_liver_b / Fu_blood;

# Concentrations in other body compartments
C_central = Q_central / V_central;
C_periph  = Q_periph  / V_periph;

# Dissolution rate
stomach_K_diss  =  K_diss * (Csat_stom  - C_stom_lu_d);
duodenum_K_diss =  K_diss * (Csat_duod  - C_duod_lu_d);
jejunum_K_diss  =  K_diss * (Csat_jeju  - C_jeju_lu_d);
ileon_K_diss    =  K_diss * (Csat_ileum - C_ileum_lu_d);
cecum_K_diss    =  K_diss * (Csat_cecum - C_cecum_lu_d);
colon_K_diss    =  K_diss * (Csat_colon - C_colon_lu_d);

# Dissolution rate in lumina
Rate_stl_diss  = stomach_K_diss  * Q_stom_lu_u;
Rate_dul_diss  = duodenum_K_diss * Q_duod_lu_u;
Rate_jel_diss  = jejunum_K_diss  * Q_jeju_lu_u;
Rate_ill_diss  = ileon_K_diss    * Q_ileum_lu_u;
Rate_cel_diss  = cecum_K_diss    * Q_cecum_lu_u;
Rate_col_diss  = colon_K_diss    * Q_colon_lu_u;
```

```
# Precipitations rate in lumina
Rate_stl_precip = K_precip * Q_stom_lu_d;
Rate_dul_precip = K_precip * Q_duod_lu_d;
Rate_jel_precip = K_precip * Q_jeju_lu_d;
Rate_ill_precip = K_precip * Q_ileum_lu_d;
Rate_cel_precip = K_precip * Q_cecum_lu_d;
Rate_col_precip = K_precip * Q_colon_lu_d;

# Transfers from lumen to lumen or feces (intestinal transit) (dissolved)
Rate_stl2dul = Kt_stom  * C_stom_lu_d;
Rate_dul2jel = Kt_duod  * C_duod_lu_d;
Rate_jel2ill = Kt_jeju  * C_jeju_lu_d;
Rate_ill2cel = Kt_ileum * C_ileum_lu_d;
Rate_cel2col = Kt_cecum * C_cecum_lu_d;
Rate_col2fel = Kt_colon * C_colon_lu_d;

# Transfers from lumen to lumen or feces (intestinal transit) (undissolved)
Rate_stu2duu = Kt_stom  * C_stom_lu_u;
Rate_duu2jeu = Kt_duod  * C_duod_lu_u;
Rate_jeu2ilu = Kt_jeju  * C_jeju_lu_u;
Rate_ilu2ceu = Kt_ileum * C_ileum_lu_u;
Rate_ceu2cou = Kt_cecum * C_cecum_lu_u;
Rate_cou2feu = Kt_colon * C_colon_lu_u;

# Transfers from lumina to epithelia for dissolved form: pasive, active
# influx and active efflux
Rate_stl2ste = Ka_stom  * (C_stom_lu_d  - Ke_over_a_epit * Cu_stom_ep) +
               Vmax_inf_stom * C_stom_lu_d / (Km_inf + C_stom_lu_d) -
               Vmax_eff_stom * Cu_stom_ep  / (Km_eff + Cu_stom_ep);
Rate_dul2due = Ka_duod  * (C_duod_lu_d  - Ke_over_a_epit * Cu_duod_ep) +
               Vmax_inf_duod * C_duod_lu_d / (Km_inf + C_duod_lu_d) -
               Vmax_eff_duod * Cu_duod_ep  / (Km_eff + Cu_duod_ep);
Rate_jel2jee = Ka_jeju  * (C_jeju_lu_d  - Ke_over_a_epit * Cu_jeju_ep) +
               Vmax_inf_jeju * C_jeju_lu_d / (Km_inf + C_jeju_lu_d) -
               Vmax_eff_jeju * Cu_jeju_ep  / (Km_eff + Cu_jeju_ep);
Rate_ill2ile = Ka_ileum * (C_ileum_lu_d - Ke_over_a_epit * Cu_ileum_ep) +
               Vmax_inf_ileum * C_ileum_lu_d / (Km_inf + C_ileum_lu_d) -
               Vmax_eff_ileum * Cu_ileum_ep  / (Km_eff + Cu_ileum_ep);
Rate_cel2cee = Ka_cecum * (C_cecum_lu_d - Ke_over_a_epit * Cu_cecum_ep) +
               Vmax_inf_cecum * C_cecum_lu_d / (Km_inf + C_cecum_lu_d) -
               Vmax_eff_cecum * Cu_cecum_ep  / (Km_eff + Cu_cecum_ep);
Rate_col2coe = Ka_colon * (C_colon_lu_d - Ke_over_a_epit * Cu_colon_ep) +
               Vmax_inf_colon * C_colon_lu_d / (Km_inf + C_colon_lu_d) -
               Vmax_eff_colon * Cu_colon_ep  / (Km_eff + Cu_colon_ep);

Rate_blo2liv = Vmax_inf_liver * Cu_liver_b / (Km_inf + Cu_liver_b) -
               Vmax_eff_liver * Cu_liver   / (Km_eff + Cu_liver);
```

```
# Transfers from epithelia to wall tissues
Rate_ste2stw = Ka_stom  * (Cu_stom_ep  - Ke_over_a_tiss * Cu_stom_w);
Rate_due2duw = Ka_duod  * (Cu_duod_ep  - Ke_over_a_tiss * Cu_duod_w);
Rate_jee2jew = Ka_jeju  * (Cu_jeju_ep  - Ke_over_a_tiss * Cu_jeju_w);
Rate_ile2ilw = Ka_ileum * (Cu_ileum_ep - Ke_over_a_tiss * Cu_ileum_w);
Rate_cee2cew = Ka_cecum * (Cu_cecum_ep - Ke_over_a_tiss * Cu_cecum_w);
Rate_coe2cow = Ka_colon * (Cu_colon_ep - Ke_over_a_tiss * Cu_colon_w);

# Transfers from tissues wall to portal vein
Rate_stw2b = Flow_stom  * C_stom_b;
Rate_duw2b = Flow_duod  * C_duod_b;
Rate_jew2b = Flow_jeju  * C_jeju_b;
Rate_ilw2b = Flow_ileum * C_ileum_b;
Rate_cew2b = Flow_cecum * C_cecum_b;
Rate_cow2b = Flow_colon * C_colon_b;

# Portal rate in
RateIn_portvein = Rate_stw2b + Rate_duw2b + Rate_jew2b +
                  Rate_ilw2b + Rate_cew2b + Rate_cow2b;

# Differential equations

# Quantity released to the stomach in case of delayed release
dt (Q_released) = Release_rate;

# Quantity to be released in the stomac in case of linear delayed release
# dt (Q_to_release) = Oral_dose_rate - Release_rate;

# Quantities absorbed
dt (Q_absorb_stom)  = Rate_stl2ste;
dt (Q_absorb_duod)  = Rate_dul2due;
dt (Q_absorb_jeju)  = Rate_jel2jee;
dt (Q_absorb_ileum) = Rate_ill2ile;
dt (Q_absorb_cecum) = Rate_cel2cee;
dt (Q_absorb_colon) = Rate_col2coe;

# Quantity reaching the portal vein
dt (Q_absorb_pv) = RateIn_portvein;

# Elimination and metabolism

dt (Q_elim_central) = Kelim * Q_central;

dt (Q_met_stom)  = Vmax_met_stom  * Cu_stom_ep /
                   (Km_met * Fu_vitro_stom + Cu_stom_ep);

dt (Q_met_duod)  = Vmax_met_duod  * Cu_duod_ep /
                   (Km_met * Fu_vitro_duod + Cu_duod_ep);
```

```
dt (Q_met_jeju)  = Vmax_met_jeju  * Cu_jeju_ep /
                   (Km_met * Fu_vitro_jeju + Cu_jeju_ep);

dt (Q_met_ileum) = Vmax_met_ileum * Cu_ileum_ep /
                   (Km_met * Fu_vitro_ileum + Cu_ileum_ep);

dt (Q_met_cecum) = Vmax_met_cecum * Cu_cecum_ep /
                   (Km_met * Fu_vitro_cecum + Cu_cecum_ep);

dt (Q_met_colon) = Vmax_met_colon * Cu_colon_ep /
                   (Km_met * Fu_vitro_colon + Cu_colon_ep);

dt (Q_met_liver) = Vmax_met_liver * Cu_liver /
                   (Km_met * Fu_vitro_liver + Cu_liver);

dt (Q_feces_d) = Rate_col2fel;

dt (Q_feces_u) = Rate_cou2feu;

# Distributions

dt(Q_periph)  = Kc2p * Q_central - Kp2c * Q_periph;

Q_out_liver   = (Flow_portvein + Flow_art_liv) * C_liver_b;

dt(Q_central) = IV_dose_rate + Q_out_liver - Rate_blo2liv -
                (Flow_portvein + Flow_art_liv) * C_central -
                dt(Q_periph) - dt(Q_elim_central);

dt(Q_liver)   = Flow_art_liv * C_central + RateIn_portvein -
                Q_out_liver + Rate_blo2liv -
                dt(Q_met_liver);

# Stomach
dt(Q_stom_lu_d) = (G_immediate_d > 0.5 ? Oral_dose_rate :
                    (G_delayed_d > 0.5 ? Release_rate : 0.0)) -
                  Rate_stl2dul - Rate_stl2ste +
                  Rate_stl_diss - Rate_stl_precip;

dt(Q_stom_lu_u) = (G_immediate_u > 0.5 ? Oral_dose_rate :
                    (G_delayed_u > 0.5 ? Release_rate : 0.0)) -
                  Rate_stu2duu - Rate_stl_diss  + Rate_stl_precip;

dt(Q_stom_ep)   = Rate_stl2ste - Rate_ste2stw - dt (Q_met_stom);

dt(Q_stom_w)    = Flow_stom * C_central + Rate_ste2stw - Rate_stw2b;
```

```
# Duodenum
dt(Q_duod_lu_d) = Rate_stl2dul  - Rate_dul2jel - Rate_dul2due +
                  Rate_dul_diss - Rate_dul_precip;

dt(Q_duod_lu_u) = Rate_stu2duu  - Rate_duu2jeu -
                  Rate_dul_diss  + Rate_dul_precip;

dt(Q_duod_ep)   = Rate_dul2due - Rate_due2duw - dt (Q_met_duod);

dt(Q_duod_w)    = Flow_duod * C_central + Rate_due2duw - Rate_duw2b;

# Jejunum
dt(Q_jeju_lu_d) = Rate_dul2jel  - Rate_jel2ill - Rate_jel2jee +
                  Rate_jel_diss - Rate_jel_precip;

dt(Q_jeju_lu_u) = Rate_duu2jeu  - Rate_jeu2ilu -
                  Rate_jel_diss + Rate_jel_precip;

dt(Q_jeju_ep)   = Rate_jel2jee - Rate_jee2jew - dt (Q_met_jeju);

dt(Q_jeju_w)    = Flow_jeju * C_central + Rate_jee2jew - Rate_jew2b;

# Ileum
dt(Q_ileum_lu_d) = Rate_jel2ill  - Rate_ill2cel - Rate_ill2ile +
                   Rate_ill_diss - Rate_ill_precip;

dt(Q_ileum_lu_u) = Rate_jeu2ilu  - Rate_ilu2ceu -
                   Rate_ill_diss + Rate_ill_precip;

dt(Q_ileum_ep)   = Rate_ill2ile - Rate_ile2ilw - dt (Q_met_ileum);

dt(Q_ileum_w)    = Flow_ileum * C_central + Rate_ile2ilw - Rate_ilw2b;

# Cecum
dt(Q_cecum_lu_d) = Rate_ill2cel  - Rate_cel2col - Rate_cel2cee +
                   Rate_cel_diss - Rate_cel_precip;

dt(Q_cecum_lu_u) = Rate_ilu2ceu  - Rate_ceu2cou -
                   Rate_cel_diss + Rate_cel_precip;

dt(Q_cecum_ep)   = Rate_cel2cee - Rate_cee2cew - dt (Q_met_cecum);

dt(Q_cecum_w)    = Flow_cecum * C_central + Rate_cee2cew - Rate_cew2b;

# Colon
dt(Q_colon_lu_d) = Rate_cel2col  - dt (Q_feces_d) - Rate_col2coe +
                   Rate_col_diss - Rate_col_precip;
```

```
  dt(Q_colon_lu_u) = Rate_ceu2cou  - dt (Q_feces_u) -
                     Rate_col_diss + Rate_col_precip;

  dt(Q_colon_ep)   = Rate_col2coe - Rate_coe2cow - dt (Q_met_colon);

  dt(Q_colon_w)    = Flow_colon * C_central + Rate_coe2cow - Rate_cow2b;

}

CalcOutputs {

  # Extra quantities and mass balance checking

  Q_gi_lu_d = Q_stom_lu_d  + Q_duod_lu_d + Q_jeju_lu_d + Q_ileum_lu_d +
              Q_cecum_lu_d + Q_colon_lu_d;

  Q_gi_lu_u = Q_stom_lu_u  + Q_duod_lu_u + Q_jeju_lu_u + Q_ileum_lu_u +
              Q_cecum_lu_u + Q_colon_lu_u;

  Q_gi_ep   = Q_stom_ep  + Q_duod_ep + Q_jeju_ep + Q_ileum_ep +
              Q_cecum_ep + Q_colon_ep;

  Q_gi_w    = Q_stom_w  + Q_duod_w + Q_jeju_w + Q_ileum_w +
              Q_cecum_w + Q_colon_w;

  Q_absorb  = Q_absorb_stom  + Q_absorb_duod  +
              Q_absorb_jeju  + Q_absorb_ileum +
              Q_absorb_cecum + Q_absorb_colon;

  Q_met_gi  = Q_met_stom  + Q_met_duod  + Q_met_jeju +
              Q_met_ileum + Q_met_cecum + Q_met_colon;

  Q_elim_gi = Q_met_gi + Q_feces_d + Q_feces_u;

  Q_elim    = Q_met_liver + Q_elim_central + Q_elim_gi;

  Q_organ   = Q_central + Q_periph + Q_liver + Q_gi_lu_d + Q_gi_ep + Q_gi_w;

  Q_total   = Q_organ + Q_elim;

  Instant_Ka_gi = (Q_gi_lu_d > 0 ? Q_absorb_pv / Q_gi_lu_d : 0);
}
End.
```

PBPK openCAT absorption model

```
# ============================================================================
# Six-compartment PBPK model with openCAT gut model.
# No difference between arterial and venous blood, no inhalation.
# Transport to tissues is flow limited.
#
# Effective permeability is assumed to be the same for each segment
# of the gi tract
#
# version 1.
#
# Units:
# - time in hours
# - volumes in liters
# - masses of substances in micromoles
# - concentrations of substances in microM
#
# Frederic Bois - Nov 2016
# ============================================================================

States  = {Q_kid,          # Quantity in kidney (micromoles)
           Q_rpt,          # ~        richly perfused tissues (viscera)
           Q_liver,        # ~        liver
           Q_adip,         # ~        adipose tissue
           Q_ppt,          # ~        poorly perfused tissues (muscle and skin)
           Q_blood,        # ~        blood
           Q_stom_lu,      # ~        stomach  lumen
           Q_duod_lu,      # ~        duodenum ~
           Q_jeju_lu,      # ~        jejunum  ~
           Q_ileum_lu,     # ~        ileum    ~
           Q_cecum_lu,     # ~        cecum    ~
           Q_colon_lu,     # ~        colon    ~
           Q_stom_ep,      # ~        stomach epithelium
           Q_duod_ep,
           Q_jeju_ep,
           Q_ileum_ep,
           Q_cecum_ep,
           Q_colon_ep,
           Q_stom_w,       # ~        stomach wall
           Q_duod_w,
           Q_jeju_w,
           Q_ileum_w,
           Q_cecum_w,
           Q_colon_w,
           Q_absorb_stom,  # ~        absorbed from stomach lumen by epithelium
           Q_absorb_duod,
```

```
            Q_absorb_jeju,
            Q_absorb_ileum,
            Q_absorb_cecum,
            Q_absorb_colon,
            Q_absorb_pv,    # ~        reaching the portal vein
            Q_elim_kid,     # ~        urine excreted or metabolized in kidney
            Q_met_liver,    # ~        metabolized in liver
            Q_met_stom,
            Q_met_duod,
            Q_met_jeju,
            Q_met_ileum,
            Q_met_cecum,
            Q_met_colon,
            Q_feces,        # ~        excreted in feces
            AUC_blood,      # Integral of blood concentration
            AUC_liver};     # ~           liver ~

Outputs = {C_plasma,        # Concentration in blood plasma (microM)
            C_kid,          # ~                kidney
            C_rpt,
            C_liver,
            C_adip,
            C_ppt,
            C_blood,
            C_stom_lu,
            C_duod_lu,
            C_jeju_lu,
            C_ileum_lu,
            C_cecum_lu,
            C_colon_lu,
            C_stom_ep,
            C_duod_ep,
            C_jeju_ep,
            C_ileum_ep,
            C_cecum_ep,
            C_colon_ep,
            C_stom_w,
            C_duod_w,
            C_jeju_w,
            C_ileum_w,
            C_cecum_w,
            C_colon_w,
            Q_gi_lu,        # Quantity in g.i. tract lumen
            Q_gi_ep,        # ~                      epithelium
            Q_gi_w,         # ~                      wall
            Q_absorb,       # ~ absorbed by the g.i. tract
            Q_elim,         # ~ metabolized or excreted
            Q_elim_gi,      # ~ metabolized or excreted from the g.i. tract only
```

```
           Q_organ,        # ~ in the body (including lumina)
           Q_total,        # ~ in the body and metabolized or excreted
           Instant_Ka_gi}; # Instantaenous net absorption ratio by g.i. tract

Inputs  = {Oral_dose_rate};# in micromoles / hours


# ------------------------
# Physiological parameters
# ------------------------

# Body mass (kg)
BDM = 70;

# Total blood flow scaling coefficient to body mass
sc_F_total = 15;

# Fractions of total blood flow going to tissues (ref Perdaems 2010)
f_Flow_adip      = 0.06;
f_Flow_kid       = 0.21;
f_Flow_ppt       = 0.20;
f_Flow_liver_art = 0.077;
f_Flow_stom      = 0.024;
f_Flow_duod      = 0.016;
f_Flow_jeju      = 0.056;
f_Flow_ileum     = 0.033;
f_Flow_cecum     = 0.006;
f_Flow_colon     = 0.038;

# Tissue fractions of body mass (ref Perdaems 2010)
f_BDM_adip  = 0.16;
f_BDM_blood = 0.065;
f_BDM_kid   = 0.0096;
f_BDM_rpt   = 0.1;
f_BDM_liver = 0.0243;
f_BDM_stom  = 0.0021;
f_BDM_duod  = 0.0003;
f_BDM_jeju  = 0.0009;
f_BDM_ileum = 0.0006;
f_BDM_cecum = 0.0005;
f_BDM_colon = 0.0048;

# Lumina as fractions of body mass (ref Perdaems 2010)
f_BDM_stom_lu  = 0.0036;
f_BDM_duod_lu  = 0.0003;
f_BDM_jeju_lu  = 0.0023;
f_BDM_ileum_lu = 0.0032;
f_BDM_cecum_lu = 0.0001;
```

```
f_BDM_colon_lu = 0.0051;

# Lengths of gi tract segments (decimeters, dm)
Length_stom  = 2.83;  # Ando 2015
Length_duod  = 1.41;  # Ando 2015
Length_jeju  = 11.68; # Ando 2015
Length_ileum = 17.52; # Ando 2015
Length_cecum = 1.7;   # Perdaems 2010
Length_colon = 11.0;  # ICRP 2002

# Radii of gi tract segments (dm)
Radius_stom  = 0.967; # Ando 2015
Radius_duod  = 0.153; # Ando 2015
Radius_jeju  = 0.137; # mean of the two Ando values
Radius_ileum = 0.098; # mean of the three Ando values
Radius_cecum = 0.35;  # Perdaems 2010
Radius_colon = 0.25;  # Perdaems 2010

# Transit half-lives in lumina (ref Perdaems 2010) (hours)
T12_stom_lu  = 0.25;
T12_duod_lu  = 0.25;
T12_jeju_lu  = 1.02;
T12_ileum_lu = 2.04;
T12_cecum_lu = 4.55;
T12_colon_lu = 13.5;

# pH of luminal contents (ref Perdaems 2010) (unused for now)
PH_stom  = 1.7;
PH_duod  = 6;
PH_jeju  = 6.5;
PH_ileum = 7.4;
PH_cecum = 5.9;
PH_colon = 7;

# Microsomal proteins (mg / g of tissue)
MicroProt_liver  = 45;
MicroProt_stom   = 0;     # to check in Houston
MicroProt_duod   = 9.45;
MicroProt_jeju   = 8.82;
MicroProt_ileum  = 1.62;
MicroProt_cecum  = 0.0;
MicroProt_colon  = 0.0;

# G.i. tract epithelial thickness (dm)
H_ep = 5.25E-4;


#-------------------------------------
```

```
# Substance-specific parameters
#-------------------------------------

# Molecular mass (g/mol)
MM;

# Administration lagtime, if needed
Tlag;

# Absorption on(1)/off(0) switches
f_Abs_stom;
f_Abs_duod;
f_Abs_jeju;
f_Abs_ileum;
f_Abs_cecum;
f_Abs_colon;

# Effective permeability of g.i. tract epithelia
Peff;

# Tissue / blood partition coefficients
PC_adip  = 1;
PC_rpt   = 1;
PC_ppt   = 1;
PC_kid   = 1;
PC_liver = 1;
PC_stom  = 1;
PC_duod  = 1;
PC_jeju  = 1;
PC_ileum = 1;
PC_cecum = 1;
PC_colon = 1;

# Plasma / blood concentration ratio
Ratio_BP = 1;

# Fractions unbound
Fu_plasma = 1.0;

Fu_met_liver = 1.0; # to liver in vitro metabolic system (e.g. microsomes)
Fu_met_stom  = 1.0; # to stomach  ~
Fu_met_duod  = 1.0; # etc.
Fu_met_jeju  = 1.0;
Fu_met_ileum = 1.0;
Fu_met_cecum = 1.0;
Fu_met_colon = 1.0;

# Michaelis-Menten Vmax (micromol/min/mg microsomal proteins)
```

```
Vmax_vitro;

# Michaelis-Menten Km (microM)
Km_vitro;

# Renal elimination clearance (L/h)
Ke_kid;


#-------------------------------------
# Scaled parameters
#-------------------------------------

# Blood flows
F_total;
f_Flow_portvein;

# Volumes
V_adip;
V_blood;
V_kid;
V_liver;
V_rpt;
V_ppt;
V_stom;
V_duod;
V_jeju;
V_ileum;
V_cecum;
V_colon;

# Tissue volumes in gi tract
V_stom_w;
V_duod_w;
V_jeju_w;
V_ileum_w;
V_cecum_w;
V_colon_w;

# Luminal volumes in gi tract
V_stom_lu;
V_colon_lu;
V_duod_lu;
V_jeju_lu;
V_ileum_lu;
V_cecum_lu;

# Epithelial volumes in gi tract
```

```
V_stom_ep;
V_duod_ep;
V_jeju_ep;
V_ileum_ep;
V_cecum_ep;
V_colon_ep;

# Epithelial surface area in gi tract (dm^2)
SA_stom;
SA_duod;
SA_jeju;
SA_ileum;
SA_cecum;
SA_colon;

# Blood flows
Flow_adip;
Flow_kid;
Flow_liver;
Flow_liver_art;
Flow_rpt;
Flow_ppt;
Flow_portvein;
Flow_stom;
Flow_duod;
Flow_jeju;
Flow_ileum;
Flow_cecum;
Flow_colon;

# GI tract absorption flows
Ka_stom;
Ka_duod;
Ka_jeju;
Ka_ileum;
Ka_cecum;
Ka_colon;

# Fraction unbound in blood
Fu_blood;

# In vivo liver and gi tract Vmax
Vmax_vivo_liver;
Vmax_vivo_stom;
Vmax_vivo_duod;
Vmax_vivo_jeju;
Vmax_vivo_ileum;
Vmax_vivo_cecum;
```

```
Vmax_vivo_colon;

# In vivo Km
Km_vivo;


Initialize {

  # Total blood flow
  F_total = sc_F_total * pow(BDM, 0.75);

  # Volumes
  V_adip     = BDM * f_BDM_adip;
  V_blood    = BDM * f_BDM_blood;
  V_kid      = BDM * f_BDM_kid;
  V_liver    = BDM * f_BDM_liver;
  V_rpt      = BDM * f_BDM_rpt;
  V_stom_w   = BDM * f_BDM_stom;
  V_duod_w   = BDM * f_BDM_duod;
  V_jeju_w   = BDM * f_BDM_jeju;
  V_ileum_w  = BDM * f_BDM_ileum;
  V_cecum_w  = BDM * f_BDM_cecum;
  V_colon_w  = BDM * f_BDM_colon;

  V_ppt = BDM * (1 - f_BDM_adip  - f_BDM_blood - f_BDM_kid -
                     f_BDM_liver - f_BDM_rpt   -
                     f_BDM_stom  - f_BDM_duod  - f_BDM_jeju -
                     f_BDM_ileum - f_BDM_cecum - f_BDM_colon);

  # Fraction of total blood flow to portal vein
  f_Flow_portvein = f_Flow_stom  + f_Flow_duod  + f_Flow_jeju +
                    f_Flow_ileum + f_Flow_cecum + f_Flow_colon;

  # Blood flows
  Flow_adip       = F_total * f_Flow_adip;
  Flow_kid        = F_total * f_Flow_kid;
  Flow_ppt        = F_total * f_Flow_ppt;
  Flow_liver_art  = F_total * f_Flow_liver_art;
  Flow_portvein   = F_total * f_Flow_portvein;
  Flow_stom       = F_total * f_Flow_stom;
  Flow_duod       = F_total * f_Flow_duod;
  Flow_jeju       = F_total * f_Flow_jeju;
  Flow_ileum      = F_total * f_Flow_ileum;
  Flow_cecum      = F_total * f_Flow_cecum;
  Flow_colon      = F_total * f_Flow_colon;

  Flow_liver      = Flow_liver_art + Flow_portvein;
```

```
Flow_rpt = F_total - Flow_adip - Flow_kid - Flow_ppt - Flow_liver;

# Epithelial surface areas
SA_stom   = Length_stom  * 2 * 3.1416 * Radius_stom;
SA_duod   = Length_duod  * 2 * 3.1416 * Radius_duod;
SA_jeju   = Length_jeju  * 2 * 3.1416 * Radius_jeju;
SA_ileum  = Length_ileum * 2 * 3.1416 * Radius_ileum;
SA_cecum  = Length_cecum * 2 * 3.1416 * Radius_cecum;
SA_colon  = Length_colon * 2 * 3.1416 * Radius_colon;

# Luminal volumes
V_stom_lu  = f_BDM_stom_lu  * BDM;
V_duod_lu  = f_BDM_duod_lu  * BDM;
V_jeju_lu  = f_BDM_jeju_lu  * BDM;
V_ileum_lu = f_BDM_ileum_lu * BDM;
V_cecum_lu = f_BDM_cecum_lu * BDM;
V_colon_lu = f_BDM_colon_lu * BDM;

# Epithelial volumes
V_stom_ep  = H_ep * SA_stom;
V_duod_ep  = H_ep * SA_duod;
V_jeju_ep  = H_ep * SA_jeju;
V_ileum_ep = H_ep * SA_ileum;
V_cecum_ep = H_ep * SA_cecum;
V_colon_ep = H_ep * SA_colon;

# Fraction unbound in blood
Fu_blood = Fu_plasma / Ratio_BP;

# Metabolic parameters, scaled from in vitro values
Vmax_vivo_liver = Vmax_vitro * 60 * MicroProt_liver * V_liver    * 1000;
Vmax_vivo_stom  = Vmax_vitro * 60 * MicroProt_stom  * V_stom_ep  * 1000;
Vmax_vivo_duod  = Vmax_vitro * 60 * MicroProt_duod  * V_duod_ep  * 1000;
Vmax_vivo_jeju  = Vmax_vitro * 60 * MicroProt_jeju  * V_jeju_ep  * 1000;
Vmax_vivo_ileum = Vmax_vitro * 60 * MicroProt_ileum * V_ileum_ep * 1000;
Vmax_vivo_cecum = Vmax_vitro * 60 * MicroProt_cecum * V_cecum_ep * 1000;
Vmax_vivo_colon = Vmax_vitro * 60 * MicroProt_colon * V_colon_ep * 1000;

Km_vivo = Km_vitro;

# GI tract absorption flows
Ka_stom  = Peff * SA_stom  * f_Abs_stom;
Ka_duod  = Peff * SA_duod  * f_Abs_duod;
Ka_jeju  = Peff * SA_jeju  * f_Abs_jeju;
Ka_ileum = Peff * SA_ileum * f_Abs_ileum;
Ka_cecum = Peff * SA_cecum * f_Abs_cecum;
Ka_colon = Peff * SA_colon * f_Abs_colon;
```

```
} # End of model scaling


Dynamics {

  # Concentrations in compartments
  C_adip    = Q_adip    / V_adip;
  C_kid     = Q_kid     / V_kid;
  C_rpt     = Q_rpt     / V_rpt;
  C_ppt     = Q_ppt     / V_ppt;
  C_liver   = Q_liver   / V_liver;
  C_stom_w  = Q_stom_w  / V_stom_w;
  C_duod_w  = Q_duod_w  / V_duod_w;
  C_jeju_w  = Q_jeju_w  / V_jeju_w;
  C_ileum_w = Q_ileum_w / V_ileum_w;
  C_cecum_w = Q_cecum_w / V_cecum_w;
  C_colon_w = Q_colon_w / V_colon_w;

  # Blood concentration
  C_blood = Q_blood / V_blood;

  # Blood concentration at compartment exit
  Ctb_kid     = C_kid     / PC_kid;
  Ctb_liver   = C_liver   / PC_liver;
  Ctb_rpt     = C_rpt     / PC_rpt;
  Ctb_ppt     = C_ppt     / PC_ppt;
  Ctb_adip    = C_adip    / PC_adip;
  Ctb_stom_w  = C_stom_w  / PC_stom;
  Ctb_duod_w  = C_duod_w  / PC_duod;
  Ctb_jeju_w  = C_jeju_w  / PC_jeju;
  Ctb_ileum_w = C_ileum_w / PC_ileum;
  Ctb_cecum_w = C_cecum_w / PC_cecum;
  Ctb_colon_w = C_colon_w / PC_colon;

  # Concentrations in lumina
  C_stom_lu  = Q_stom_lu  / V_stom_lu;
  C_duod_lu  = Q_duod_lu  / V_duod_lu;
  C_jeju_lu  = Q_jeju_lu  / V_jeju_lu;
  C_ileum_lu = Q_ileum_lu / V_ileum_lu;
  C_cecum_lu = Q_cecum_lu / V_cecum_lu;
  C_colon_lu = Q_colon_lu / V_colon_lu;

  # Concentrations in epithelia
  C_stom_ep  = Q_stom_ep  / V_stom_ep;
  C_duod_ep  = Q_duod_ep  / V_duod_ep;
  C_jeju_ep  = Q_jeju_ep  / V_jeju_ep;
  C_ileum_ep = Q_ileum_ep / V_ileum_ep;
  C_cecum_ep = Q_cecum_ep / V_cecum_ep;
```

```
C_colon_ep = Q_colon_ep / V_colon_ep;

# Transfers from lumen to lumen or feces (intestinal transit)
Rate_stl2dul = (log(2.0) * V_stom_lu  / T12_stom_lu)  * C_stom_lu;
Rate_dul2jel = (log(2.0) * V_duod_lu  / T12_duod_lu)  * C_duod_lu;
Rate_jel2ill = (log(2.0) * V_jeju_lu  / T12_jeju_lu)  * C_jeju_lu;
Rate_ill2cel = (log(2.0) * V_ileum_lu / T12_ileum_lu) * C_ileum_lu;
Rate_cel2col = (log(2.0) * V_cecum_lu / T12_cecum_lu) * C_cecum_lu;
Rate_col2fel = (log(2.0) * V_colon_lu / T12_colon_lu) * C_colon_lu;

# Transfers from lumina to epithelia
Rate_stl2ste = Ka_stom  * (C_stom_lu  - C_stom_ep);
Rate_dul2due = Ka_duod  * (C_duod_lu  - C_duod_ep);
Rate_jel2jee = Ka_jeju  * (C_jeju_lu  - C_jeju_ep);
Rate_ill2ile = Ka_ileum * (C_ileum_lu - C_ileum_ep);
Rate_cel2cee = Ka_cecum * (C_cecum_lu - C_cecum_ep);
Rate_col2coe = Ka_colon * (C_colon_lu - C_colon_ep);

# Transfers from epithelia to tissues
Rate_ste2stw = Ka_stom  * (C_stom_ep  - C_stom_w);
Rate_due2duw = Ka_duod  * (C_duod_ep  - C_duod_w);
Rate_jee2jew = Ka_jeju  * (C_jeju_ep  - C_jeju_w);
Rate_ile2ilw = Ka_ileum * (C_ileum_ep - C_ileum_w);
Rate_cee2cew = Ka_cecum * (C_cecum_ep - C_cecum_w);
Rate_coe2cow = Ka_colon * (C_colon_ep - C_colon_w);

# Transfers from tissues to portal vein
Rate_stw2pv = Flow_stom  * Ctb_stom_w;
Rate_duw2pv = Flow_duod  * Ctb_duod_w;
Rate_jew2pv = Flow_jeju  * Ctb_jeju_w;
Rate_ilw2pv = Flow_ileum * Ctb_ileum_w;
Rate_cew2pv = Flow_cecum * Ctb_cecum_w;
Rate_cow2pv = Flow_colon * Ctb_colon_w;

# Portal rate in
RateIn_portvein = Rate_stw2pv + Rate_duw2pv + Rate_jew2pv +
                  Rate_ilw2pv + Rate_cew2pv + Rate_cow2pv;

Ctb_portvein = RateIn_portvein / Flow_portvein;


# Differential equations

# Quantities absorbed
dt (Q_absorb_stom)  = Rate_stl2ste;
dt (Q_absorb_duod)  = Rate_dul2due;
dt (Q_absorb_jeju)  = Rate_jel2jee;
dt (Q_absorb_ileum) = Rate_ill2ile;
```

```
dt (Q_absorb_cecum) = Rate_cel2cee;
dt (Q_absorb_colon) = Rate_col2coe;


# Quantity reaching the portal vein
dt (Q_absorb_pv) = RateIn_portvein;


# Elimination and metabolism

dt (Q_elim_kid)  = Ke_kid * Ctb_kid * Fu_blood;

dt (Q_met_liver) = Vmax_vivo_liver * Ctb_liver  /
                   (Km_vivo * Fu_met_liver / Fu_blood + Ctb_liver);

dt (Q_met_stom)  = Vmax_vivo_stom  * (C_stom_ep / PC_stom) /
                   (Km_vivo * Fu_met_stom / Fu_blood +
                    C_stom_ep / PC_stom);

dt (Q_met_duod)  = Vmax_vivo_duod  * (C_duod_ep / PC_duod) /
                   (Km_vivo * Fu_met_duod / Fu_blood +
                    C_duod_ep / PC_duod);

dt (Q_met_jeju)  = Vmax_vivo_jeju * (C_jeju_ep / PC_jeju) /
                   (Km_vivo * Fu_met_jeju / Fu_blood +
                    C_jeju_ep / PC_jeju);

dt (Q_met_ileum) = Vmax_vivo_ileum * (C_ileum_ep / PC_ileum) /
                   (Km_vivo * Fu_met_ileum / Fu_blood +
                    C_ileum_ep / PC_ileum);

dt (Q_met_cecum) = Vmax_vivo_cecum * (C_cecum_ep / PC_cecum) /
                   (Km_vivo * Fu_met_cecum / Fu_blood +
                    C_cecum_ep / PC_cecum);

dt (Q_met_colon) = Vmax_vivo_colon * (C_colon_ep / PC_colon) /
                   (Km_vivo * Fu_met_colon / Fu_blood +
                    C_colon_ep / PC_colon);

dt (Q_feces) = Rate_col2fel;


# Distributions

dt (Q_kid)    = Flow_kid  * (C_blood - Ctb_kid) - dt (Q_elim_kid);
dt (Q_adip)   = Flow_adip * (C_blood - Ctb_adip);
dt (Q_ppt)    = Flow_ppt  * (C_blood - Ctb_ppt);
dt (Q_rpt)    = Flow_rpt  * (C_blood - Ctb_rpt);
```

```
  dt (Q_liver)  = Flow_liver_art * C_blood + Flow_portvein * Ctb_portvein -
                  Flow_liver * Ctb_liver - dt (Q_met_liver);

  dt(Q_blood)   = Flow_rpt  * Ctb_rpt  + Flow_ppt   * Ctb_ppt +
                  Flow_adip * Ctb_adip + Flow_liver * Ctb_liver +
                  Flow_kid  * Ctb_kid  - F_total * C_blood;

  dt(Q_stom_lu) = Oral_dose_rate - Rate_stl2dul - Rate_stl2ste;
  dt(Q_stom_ep) = Rate_stl2ste - Rate_ste2stw - dt (Q_met_stom);
  dt(Q_stom_w)  = Flow_stom * C_blood + Rate_ste2stw - Rate_stw2pv;

  dt(Q_duod_lu) = Rate_stl2dul - Rate_dul2jel - Rate_dul2due;
  dt(Q_duod_ep) = Rate_dul2due - Rate_due2duw - dt (Q_met_duod);
  dt(Q_duod_w)  = Flow_duod * C_blood + Rate_due2duw - Rate_duw2pv;

  dt(Q_jeju_lu) = Rate_dul2jel - Rate_jel2ill - Rate_jel2jee;
  dt(Q_jeju_ep) = Rate_jel2jee - Rate_jee2jew - dt (Q_met_jeju);
  dt(Q_jeju_w)  = Flow_jeju * C_blood + Rate_jee2jew - Rate_jew2pv;

  dt(Q_ileum_lu) = Rate_jel2ill - Rate_ill2cel - Rate_ill2ile;
  dt(Q_ileum_ep) = Rate_ill2ile - Rate_ile2ilw - dt (Q_met_ileum);
  dt(Q_ileum_w)  = Flow_ileum * C_blood + Rate_ile2ilw - Rate_ilw2pv;

  dt(Q_cecum_lu) = Rate_ill2cel - Rate_cel2col - Rate_cel2cee;
  dt(Q_cecum_ep) = Rate_cel2cee - Rate_cee2cew - dt (Q_met_cecum);
  dt(Q_cecum_w)  = Flow_cecum * C_blood + Rate_cee2cew - Rate_cew2pv;

  dt(Q_colon_lu) = Rate_cel2col - Rate_col2coe - dt (Q_feces);
  dt(Q_colon_ep) = Rate_col2coe - Rate_coe2cow - dt (Q_met_colon);
  dt(Q_colon_w)  = Flow_colon * C_blood + Rate_coe2cow - Rate_cow2pv;

  # AUCs

  dt(AUC_blood) = C_blood;
  dt(AUC_liver) = C_liver;

}

CalcOutputs {

  C_plasma = (C_blood > 0 ? C_blood / Ratio_BP : 1E-10);

  # Mass balance checking

  Q_gi_lu   = Q_stom_lu  + Q_duod_lu + Q_jeju_lu + Q_ileum_lu +
              Q_cecum_lu + Q_colon_lu;

  Q_gi_ep   = Q_stom_ep  + Q_duod_ep + Q_jeju_ep + Q_ileum_ep +
```

```
              Q_cecum_ep + Q_colon_ep;

  Q_gi_w    = Q_stom_w  + Q_duod_w + Q_jeju_w + Q_ileum_w +
              Q_cecum_w + Q_colon_w;

  Q_absorb  = Q_absorb_stom  + Q_absorb_duod  +
              Q_absorb_jeju  + Q_absorb_ileum +
              Q_absorb_cecum + Q_absorb_colon;

  Q_elim_gi = Q_met_stom  + Q_met_duod  + Q_met_jeju  +
              Q_met_ileum + Q_met_cecum + Q_met_colon + Q_feces;

  Q_elim    = Q_elim_kid + Q_met_liver + Q_elim_gi;

  Q_organ   = Q_kid   + Q_rpt   + Q_liver + Q_adip + Q_ppt +
              Q_blood + Q_gi_lu + Q_gi_ep + Q_gi_w;

  Q_total   = Q_organ + Q_elim;

  Instant_Ka_gi = (Q_gi_lu > 0 ? Q_absorb_pv / Q_gi_lu : 0);

}

End..
```